


RICARDO AGUADO-MUÑOZ



# GENERACION ALEATORIA DE EJERCICIOS CON ORDENADOR

**De los mismos autores:**

**BASIC BASICO  
CURSOS DE PROGRAMACION**

R. Aguado, A. Blanco, J. Zabala y R. Zamarreño

Es un libro de texto o divulgativo, fruto de la experiencia didáctica de los autores.

Explica la programación en BASIC. Muchos de los ejercicios que propone se resuelven en el libro PROGRAMAS COMENTADOS DE BASIC BASICO.

**PROGRAMAS COMENTADOS  
DE BASIC BASICO**

R. Aguado, A. Blanco, E. Rubiales, J. Zabala y R. Zamarreño

Ciento ocho problemas de programación explicados y comentados.

Cincuenta de ellos responden a los ejercicios propuestos en el libro BASIC BASICO CURSO DE PROGRAMACION.

**BASIC JUNIOR  
INICIACION A LA PROGRAMACION**

R. Aguado, A. Blanco, E. Rubiales, J. Zabala y R. Zamarreño

Está destinado a escolares y principiantes.

Es formativo y ameno. Tiene numerosos ejemplos divertidos.

BASIC JUNIOR te invita a jugar el apasionante juego de la programación de ordenadores.

**PROGRAMAS EXPLICADOS  
DE BASIC JUNIOR**

R. Aguado, A. Blanco, E. Rubiales, J. Zabala y R. Zamarreño

Contiene resueltos y explicados los cien ejercicios propuestos en BASIC JUNIOR. INICIACION A LA PROGRAMACION.

Los programas resultan comprensibles y amenos a los más jóvenes.

**EL BASIC DEL SPECTRUM.  
DEL TECLADO AL MICRODRIVE**

A. Blanco y B. Compostela

Hace fácil el Spectrum.

Especialmente pensado para aprovechar sus posibilidades gráficas y sonoras.

Inicia de forma amena a la programación estructurada.

1400  
1489

# GENERACION ALEATORIA DE EJERCICIOS CON ORDENADOR

RICARDO AGUADO-MUÑOZ

**DISTRIBUYE:**  
**GRUPO DISTRIBUIDOR EDITORIAL, S. A.**  
Don Ramón de la Cruz, 67  
28001 MADRID  
Tfno.: 91-401 12 00

Reservados todos los derechos. Queda prohibida la reproducción total o parcial  
de esta obra sin previo consentimiento por escrito del autor

Editorial El Ordenador Amigo, c/ Villa de Marín, 41. 28029 Madrid.  
© Ricardo Aguado-Muñoz

I.S.B.N.: 84-86681-00-6

Depósito legal: M. 12.778-1987

Imprime: Hijos de E. Minuesa, S. L. Ronda de Toledo, 24. 28005 Madrid

*A Conchi, Inmaculada,  
Ricardo y Rocío*



# INDICE

<b>PRESENTACION</b> .....	9
<b>1. INTRODUCCION</b> .....	11
1.1 Generación aleatoria de ejercicios.....	11
1.2 Esquema general .....	12
1.3 ¿Resolver los ejercicios o prepararlos? .....	13
1.4 Generación aleatoria de los números .....	14
1.5 Rango de los números .....	15
1.6 Enmendando al azar .....	15
1.7 En los sorteos siempre salen los mismos premios .....	16
1.8 Estructura de los programas .....	16
1.9 El tipo de las variables .....	17
1.10 Salidas por impresora.....	18
1.11 Salidas por pantalla o por impresora.....	19
1.12 Cómo preparar un disco autoejecutable .....	22
<b>2. DIVISIBILIDAD</b> .....	25
2.1 Descomposición de un número en factores primos .....	25
2.2 Máximo común divisor y mínimo común múltiplo .....	28
<b>3. FRACCIONES</b> .....	31
3.1 Simplificación de fracciones .....	31
3.2 Operaciones con fracciones.....	35
<b>4. SISTEMAS DE NUMERACION</b> .....	41
4.1 Cambio de base.....	41
4.2 Sumar y restar en distintas bases .....	47
<b>5. ECUACIONES</b> .....	53
5.1 Ecuaciones lineales .....	53
5.2 Ecuaciones de segundo grado.....	56

6.	<b>POLINOMIOS</b> .....	61
6.1	Escritura de polinomios.....	61
6.2	Multiplicación de polinomios .....	64
6.3	División de polinomios.....	65
6.4	División de Ruffini.....	76
6.5	Valor numérico de un polinomio .....	82
6.6	Factorización de polinomios .....	85
6.7	Ecuaciones de cuarto grado con raíces enteras .....	89
6.8	Operaciones con fracciones algebraicas .....	94
7.	<b>DETERMINANTES Y SISTEMAS</b> .....	101
7.1	Determinantes.....	101
7.2	Sistemas de ecuaciones lineales.....	107
8.	<b>ESTADISTICA</b> .....	113
8.1	Distribución normal.....	113
8.2	Distribución bidimensional.....	120
9.	<b>RESOLUCION DE TRIANGULOS</b> .....	125
9.1	Triángulos rectángulos .....	125
9.2	Triángulos cualesquiera .....	131
10.	<b>REPRESENTACION GRAFICA DE FUNCIONES</b> .....	139
10.1	Representación gráfica de funciones.....	139



## PRESENTACION

El panorama de las aplicaciones de los ordenadores a la enseñanza está sobrado de discursos teóricos y falto de aplicaciones concretas. Por eso, y a pesar de tener algunas dudas sobre su alcance, me he animado a publicar estos programas de *generación aleatoria de ejercicios* (G.A.E.), con la esperanza de que resulten útiles en sí mismos, o que susciten en el lector ideas que contribuyan a abrir nuevos caminos en el resistente campo de las aplicaciones educativas de los ordenadores.

Estos programas, que han sido diseñados para ser utilizados dentro del marco de la enseñanza convencional, no suponen, ni predisponen, ninguna didáctica concreta, pues todas consideran necesaria la *ejercitación* del alumno.

Creo conveniente advertir que lo que se va a exponer en este libro, nada tiene que ver con la denominada enseñanza asistida por ordenador (E.A.O.), aunque los programas de G.A.E. bien pudieran integrarse en programas de E.A.O.

En el texto se exponen las ideas básicas sobre la generación aleatoria de ejercicios, así como observaciones útiles para el lector que desee programar por su cuenta otros ejercicios, bien sean de matemáticas, de física o de otras materias.

Este libro puede ser útil a todos aquellos que necesitan disponer de muchos ejercicios del mismo tipo y de las soluciones. En especial, a profesores, estudiantes y padres de estudiantes.

Los profesores podrán disponer de innumerables ejercicios para realizarlos en clase, proponerlos como deberes o preparar exámenes. Los estudiantes tendrán una amplia colección de enunciados en los que

ejercitarse y cotejar las soluciones inmediatamente. Los padres podrán obtener ejercicios y proponérselos a sus hijos para reforzar sus habilidades y ayudarles a preparar exámenes.

Todos los algoritmos que se presentan han sido codificados en BASIC, el lenguaje de programación más sencillo y popular. Los programas corren sin dificultad en una amplia gama de ordenadores domésticos y personales; en particular deben funcionar en los del sistema MSX y en los PC de IBM y compatibles. Al lector le será fácil realizar las escasas modificaciones que requieren los programas para ser ejecutados en otros ordenadores.

Los listados, tal como vienen en el texto, están preparados para obtener las salidas (enunciados y soluciones) por la pantalla. Aconsejo vivamente al lector que lea el capítulo 1, en donde se explica cómo arreglar los programas para que los resultados salgan siempre por la impresora, o para que el ordenador ofrezca la alternativa de salidas por la impresora o por la pantalla, según convenga.

En el capítulo 1 también se explica la manera de preparar un disco para manejar todos los programas del libro de forma cómoda y sencilla, con objeto de que las personas sin conocimientos informáticos puedan usarlos.

Mi agradecimiento a Saltés por la portada que arroja al texto, y a Agustín Blanco, Javier Zabala y Ricardo Zamarreño por las sugerencias que me han hecho llegar, después de haber leído pacientemente la versión manuscrita.

Ricardo Aguado-Muñoz

# 1

## INTRODUCCION

### 1.1 GENERACION ALEATORIA DE EJERCICIOS

Se entiende por *generación aleatoria de ejercicios* (G.A.E.), la creación por ordenador de un número prefijado de ejercicios del mismo tipo, en los que los datos de los enunciados (o las soluciones) se generan al azar. El ordenador escribe en la pantalla, o en el papel de la impresora, una lista con todos los enunciados numerados y otra con las correspondientes soluciones.

Veámoslo con un ejemplo:

Se desea generar 40 ejercicios de máximo común divisor y mínimo común múltiplo de dos números. La lista de enunciados puede ser como ésta:

- Núm. 1 Halla el m.c.d. y el m.c.m. de los números 880 y 48  
Núm. 2 Halla el m.c.d. y el m.c.m. de los números 495 y 819  
:  
:  
Núm. 40 Halla el m.c.d. y el m.c.m. de los números 891 y 66

La lista de soluciones puede ser así:

Soluciones:

- Núm. 1  
m.c.d. = 16  
m.c.m. = 2640  
  
Núm. 2  
m.c.d. = 9  
m.c.m. = 45045  
:  
Núm. 40  
m.c.d. = 33  
m.c.m. = 1782

El ordenador, para cada ejercicio, genera al azar un par de números enteros positivos  $A(K)$ ,  $B(K)$ , y escribe el enunciado:

Núm. k    Halla el m.c.d. y el m.c.m. de los números  $A(K)$  y  $B(K)$

Repite esta operación 40 veces para  $K = 1,2,...,40$ .

En una segunda fase el ordenador calcula el m.c.d. y el m.c.m. de los números  $A(K)$  y  $B(K)$ , y escribe la solución:

Núm. k


m.c.d. = D

m.c.m. = M

Repite esto 40 veces, para  $K = 1,2,...,40$ .

## 1.2 ESQUEMA GENERAL

El esquema general de todo algoritmo de G.A.E. es el siguiente:



En el ejemplo precedente, el ordenador en la primera fase obtiene aleatoriamente los datos  $A(K)$  y  $B(K)$  de cada ejercicio. En la segunda

fase calcula los resultados  $D$  y  $M$  inmediatamente antes de escribirlos en la pantalla. Observe que las variables  $D$  y  $M$  no son de índice.

Otra posibilidad hubiera sido generar los datos  $A$  y  $B$  como variables simples, escribir el enunciado y calcular las soluciones  $D(K)$  y  $M(K)$  como variables de índice, todo ello en la primera fase. En la segunda se escribirían las soluciones  $D(K)$  y  $M(K)$ , para cada valor de  $K$ .

Las variables de índice ocupan mucha memoria de ordenador, por eso hay que procurar usar las menos posibles. En el ejemplo que nos ocupa necesitamos dos: bien sean  $A(K)$  y  $B(K)$ , o bien  $D(K)$  y  $M(K)$ . Sin embargo, si se tratase de preparar ejercicios únicamente de m.c.d., es evidente que lo mejor sería utilizar las variables simples  $A$  y  $B$  y guardar en la variable de índice  $D(K)$  las soluciones.

### **1.3 ¿RESOLVER LOS EJERCICIOS O PREPARARLOS?**

En unas ocasiones será conveniente preparar el enunciado y luego resolver el ejercicio, como en el ejemplo anterior. Sin embargo, en muchos casos es mejor generar aleatoriamente la solución y después preparar un enunciado que se acomode a esa solución. Y esto por alguna de estas razones:

#### **1. El algoritmo de resolución del ejercicio es complicado**

Pensemos, por ejemplo, en la generación de ejercicios de sistemas de cuatro ecuaciones lineales con cuatro incógnitas. Aquí es más sencillo generar primero la solución y luego preparar un sistema que tenga esa solución.

#### **2. Para garantizar el tipo de solución**

Si deseamos obtener ecuaciones de segundo grado con soluciones enteras, por ejemplo, no podemos generar al azar los coeficientes de la ecuación, so pena de obtener un alto porcentaje de ellas con soluciones no enteras o, incluso, sin solución. Lo mejor es elegir las soluciones y después preparar una ecuación que las contenga.

### 3. Para garantizar el tipo de datos

Es el caso de los ejercicios de estadística que se presentan en este libro. Para generar, por ejemplo, una distribución normal de 100 datos, y obtener como ejercicio la media y la desviación típica, hay que elegir primero la media y la desviación, y luego generar la distribución.

#### 1.4 GENERACION ALEATORIA DE LOS NUMEROS

La *generación aleatoria* de los números que intervienen en los ejercicios se realiza utilizando la función *RND* del BASIC. Como es sabido, la instrucción

$$X = \text{RND}(1)$$

asigna a *X* un número decimal aleatorio comprendido entre 0 y 1:

$$0 < X < 1 \quad (*)$$

En la mayoría de los casos estos números decimales no se utilizan directamente. Hay que transformarlos, por ejemplo así:

$$Y = \text{RND}(1) * 10$$

Esta asignación introduce en *Y* un número comprendido entre 0 y 10.

$$0 < Y < 10$$

Estos números decimales *Y* tampoco son de mucha utilidad en este libro. Sólo si obtenemos su parte entera, serán interesantes para nosotros:

$$E = \text{INT}(\text{RND}(1) * 10)$$

Ahora *E* será alguno de los números enteros 0, 1, 2, ..., 9.

En general, si se desea un número aleatorio *N* comprendido entre los números enteros *A* y *B*, entonces la asignación:

$$N = A + \text{INT}(\text{RND}(1) * (B - A))$$

proporciona un número *N* mayor o igual que *A* y menor que *B*.

---

(\*) La función *RND* puede utilizar cualquier otro argumento positivo distinto del 1, pero siempre el resultado está comprendido entre 0 y 1. Por ejemplo,  $0 < \text{RND}(2) < 1$ .

## 1.5 RANGO DE LOS NUMEROS

Los números que intervienen en los ejercicios de este libro han sido elegidos de modo que pertenezcan a un rango determinado por la naturaleza del ejercicio, por la tradición escolar y por la presentación de los textos en la pantalla. Presentación que, a su vez, está condicionada por la estética y por el número de caracteres que se pueden escribir en una línea.

Así, por ejemplo, los coeficientes de los polinomios o de las ecuaciones son números enteros de una o dos cifras, excepto en aquellas circunstancias en las que es difícil controlar los resultados. Imagínese el caso de la multiplicación de dos polinomios con coeficientes de una cifra. Es improbable que el polinomio producto tenga también los coeficientes de una sola cifra.

En algunos programas de este libro se ha podido ajustar, de un modo preciso, el rango de los números para que los resultados pertenezcan a un orden prefijado; pero en otros, tales el caso de los determinantes, el ajuste ha sido experimental, es decir, se han probado los programas con distintos parámetros hasta conseguir el efecto apetecido.

## 1.6 ENMENDANDO EL AZAR

En ocasiones no basta el puro azar para obtener buenos enunciados. Veamos un ejemplo claro. Supongamos que hemos decidido generar con el ordenador una serie de ejercicios de simplificación de fracciones. Tenemos que generar el numerador  $P$  y el denominador  $Q$  de la fracción. Decidimos que estos términos han de ser números de dos o tres cifras, es decir, números entre 10 y 999.

Las instrucciones

$$\begin{aligned}P &= \text{INT}(\text{RND}(1)*990) + 10 \\Q &= \text{INT}(\text{RND}(1)*990) + 10\end{aligned}$$

nos proporcionan los términos dentro del rango fijado; pero la mayor parte de las fracciones así obtenidas son irreducibles (aproximadamente un 60 %). Son demasiados ejercicios de simplificación que ... ¡no se pueden simplificar!

Para corregir este defecto podemos proceder así: Cada vez que el m.c.d. ( $P, Q$ ) es 1, multiplicamos ambos términos de la fracción por un

número aleatorio comprendido entre 1 y 9. Con lo que reducimos la presencia de fracciones irreducibles a un 7 %, que consideramos razonable.

## 1.7 EN LOS SORTEOS SIEMPRE SALEN LOS MISMOS PREMIOS

Con esta frase se quiere llamar la atención sobre el hecho de que siempre que se ejecuta un programa de este libro, se obtiene la misma lista de ejercicios.

Hablando en general, este hecho supone una ventaja, más que un inconveniente. Imagine que ayer propuso un ejercicio a cada uno de sus 42 alumnos y que hoy no encuentra la lista de soluciones. Ninguna tragedia. Usted ejecuta de nuevo el programa y vuelve a tener los mismos 42 ejercicios con sus 42 soluciones.

Claro, que puede ocurrir que usted desee cambiar totalmente de enunciados de una vez para otra. Esto también tiene remedio: incluya en el programa las siguientes líneas:

```
15 INPUT "Número clave ";C
16 FOR I=1 TO C : Z=RND(1) : NEXT I
```

Ahora la lista de ejercicios dependerá del número clave que dé al ordenador, como contestación al INPUT de la línea 15.

La explicación es la siguiente: Cada vez que se ejecuta un programa, el ordenador genera con RND(1) la misma sucesión de números aleatorios. El efecto que producen las líneas 15 y 16 es el de desechar los  $C$  primeros números aleatorios y comenzar a utilizar la serie a partir del lugar  $C+1$ . Al variar este lugar, varían los enunciados y las soluciones.

## 1.8 ESTRUCTURA DE LOS PROGRAMAS

Todos los programas que se ofrecen en este libro tienen la misma estructura:

*Título del programa*  
*Borrado de la pantalla*  
*Nombre de los ejercicios*  
*Cuántos ejercicios? EJ*  
*(Precisiones sobre el tipo de ejercicios)*  
*Dimensionado de las variables de índice*



```
FOR K = 1 TO EJ  
Escritura del enunciado núm. K  
NEXT K
```

```
FOR K = 1 TO EJ  
Escritura de la solución núm. K  
NEXT K
```

*Fin*

En todos los programas la variable *EJ* guarda el número de ejercicios deseado y la variable *K* el número de orden del ejercicio.

Algunos programas llevan al principio lo que hemos llamado «Precisiones sobre el tipo de ejercicios». Por ejemplo, el programa *SISTEMAS.BAS* nos permite escoger entre sistemas de 2, 3, 4, 5 ó 6 ecuaciones.

## 1.9 EL TIPO DE LAS VARIABLES

Siempre que se ha considerado necesario, se han declarado las variables de *tipo entero*, incluyendo al principio de los programas la línea

DEFINIT A—Z

lo cual ahorra memoria y acelera la ejecución.

En algún caso concreto las variables enteras no se han podido declarar como tales, porque hay cálculos intermedios en los que pueden tomar valores que desbordan el intervalo  $-32768$ ,  $+32767$  que, como se sabe, es el rango de los números de tipo entero en los pequeños ordenadores. En este, y en otros casos, las variables han sido declaradas como de *simple precisión* con la línea:

DEFSNG A—Z

Por cierto, si su ordenador no dispone de esta clase de declaraciones, suprima en los programas la línea que corresponda. La declaración de una variable de tipo entero siempre la podrá hacer añadiendo a su nombre el símbolo %.

## 1.10 SALIDAS POR IMPRESORA

Para una más eficaz utilización de los programas es necesario disponer de una impresora que escriba sobre el papel los enunciados y las soluciones de los ejercicios.

Si usted maneja los programas tal como vienen en el libro, obtendrá todas las salidas por pantalla. En el caso de que desee obtener las listas de enunciados y soluciones en papel, deberá dirigir estas salidas hacia la impresora.

### Sistema operativo MS-DOS

Con el sistema MS-DOS es muy fácil dirigir las salidas de los programas hacia la impresora. Veámoslo en el caso del máximo común divisor, cuyo programa suponemos que lo tiene en el disco A:

Escriba:

```
A > basic mcdmcm > lpt1      (<RETURN>)
```

A partir de este momento, el programa MCDMCM.BAS se desarrollará en la pantalla y en la impresora, simultáneamente.

Terminada la ejecución, el ordenador se queda en el modo BASIC y con las salidas dirigidas hacia la impresora. Para salir de aquí, escriba SYSTEM o provoque un error cualquiera. Desgraciadamente las últimas líneas de la pantalla se escriben también en el papel de la impresora. Como se ve, esta manera de regresar al MS-DOS no es muy elegante.

Lo mejor es cambiar, en el programa BASIC, la instrucción END por la SYSTEM, con lo que el propio programa nos devolverá al sistema operativo. Compruébelo, por favor.

### Sistema MSX

En el BASIC del MSX (también se puede hacer en el BASIC de los compatibles) hay que sustituir todas las instrucciones PRINT (a partir de las de escritura de enunciados) por instrucciones LPRINT, es decir, hay que insertar una L delante de todos los PRINT.

Una manera de hacerlo es listar el programa e incluir manualmente la L en todas las líneas que sea necesario, sin olvidar pulsar <RETURN> cada vez que se haya arreglado una línea.

Una vez modificado el programa, compruébelo y guárdelo en el disco con un nombre parecido al original. Por ejemplo, si acaba de modificar el programa MCDMCM.BAS, puede guardar la versión para impresora con el nombre de MCDMCM.IMP.

## 1.11 SALIDAS POR PANTALLA O POR IMPRESORA

En realidad usted no necesita un programa para obtener las salidas por la pantalla y otro para obtenerlas por impresora: un solo programa basta para dirigir las salidas a uno o a otro sitio, según le convenga.

Esto ha quedado claro en el apartado anterior, cuando se utilizaba el MS-DOS. Ahora se trata de ver cómo se puede hacer desde el BASIC.

Recordemos, con un sencillo ejemplo, cómo se escriben los datos en un fichero secuencial:

```
10 REM Escritura en un fichero
20 OPEN "A:PRUEBA.DAT" FOR OUTPUT AS 1
30 PRINT # 1,"HOLA"
40 CLOSE 1
50 END
```

La línea 20 del programa abre un fichero secuencial, de nombre PRUEBA.DAT, en el dispositivo A: (en el disco que está en A:) y le asigna el número 1.

La línea 30 escribe en el fichero 1 la palabra HOLA. La 40 cierra el fichero.

La impresora funciona como un fichero en el dispositivo LPT1: . Modificando ligeramente el programa anterior, podemos hacer que el ordenador escriba en el papel de la impresora la palabra HOLA :

```
10 REM Escritura en la impresora
20 OPEN "LPT1:" FOR OUTPUT AS 1
30 PRINT # 1,"HOLA"
40 CLOSE 1
50 END
```

Observe que en la línea 20 únicamente se pone entre comillas el nombre del dispositivo "LPT1:".

En el sistema MSX, la línea 20 se escribe así:

```
20 OPEN "LPT:" AS 1
```

**La pantalla** también funciona como un fichero secuencial. Su nombre es "SCRN:" en los sistemas MS-DOS, y "CRT:" en MSX.

Modifiquemos otra vez el programa para conseguir que el ordenador escriba HOLA en la pantalla. Sólo hay que cambiar "LPT1:" por "SCRN:".

```
10 REM Escritura en la pantalla
20 OPEN "SCRN:" FOR OUTPUT AS 1
30 PRINT # 1,"HOLA"
40 CLOSE 1
50 END
```

En los ordenadores MSX, hay que poner:

```
20 OPEN "CRT:" AS 1
```

Ejecute este programa y verá que se escribe HOLA en la pantalla.

En esencia los dos últimos programas son lo mismo. Varía sólo el nombre del dispositivo. Por esto es fácil tener un único programa que escriba los datos en la pantalla o en la impresora:

```
10 REM Escritura en la pantalla o en la impresora
15 INPUT "Salidas por pantalla ( 0 ) o por
impresora ( 1 )";Z
20 IF Z=1 THEN OPEN "LPT1:" FOR OUTPUT AS 1 ELSE
OPEN "SCRN:" FOR OUTPUT AS 1
30 PRINT # 1,"HOLA"
40 CLOSE 1
50 END
```

La línea 20, en el sistema MSX, queda así:

```
20 IF Z=1 THEN OPEN "LPT:" AS 1 ELSE OPEN "CRT:" AS 1
```

Si respondemos con un 1 a la interrogación provocada por la línea 15, la palabra HOLA se escribirá en el papel de la impresora. Si respondemos con 0, entonces HOLA aparecerá en la pantalla.

Lo expuesto hasta aquí da idea de lo que hay que hacer para modificar los programas de este libro en el caso de que le interese la alternativa pantalla / impresora.

Se puede proceder así:

1. Introducir las líneas:

```
INPUT "Salidas por pantalla ( 0 ) o por  
impresora ( 1 )";Z
```

```
IF Z=1 THEN OPEN "LPT1:" FOR OUTPUT AS 1 ELSE  
OPEN "SCRN:" FOR OUTPUT AS 1
```

al principio del programa, antes de la escritura de los enunciados.

2. Insertar la línea CLOSE 1, antes de la línea END.

Otra forma de hacerlo es sustituir la palabra END por CLOSE 1 : END

3. A todos los PRINT que escriben los enunciados y soluciones, añadirles # 1, (no olvide la coma).

Con un procesador de texto, o con el editor de línea EDLIN, se puede sustituir automáticamente PRINT por PRINT # 1,.

Veamos un ejemplo concreto:

El programa MCDMCM.BAS de la página 23, puede quedar modificado de esta manera:

```
10 REM Máximo común divisor y mínimo común múltiplo  
20 CLS  
30 PRINT "MAXIMO COMUN DIVISOR Y MINIMO COMUN MULTI  
PLO"  
40 PRINT : PRINT : INPUT "Cuántos ejercicios";EJ  
45 INPUT "Salidas por pantalla ( 0 ) o por impreso  
ra ( 1 )";Z
```

```

47 IF Z=1 THEN OPEN "LPT1:" FOR OUTPUT AS 1 ELSE
OPEN "SCRN:" FOR OUTPUT AS 1
50 DIM A(EJ),B(EJ)
60 FOR K=1 TO EJ
70 A1=INT(RND(1)*13+1) : B1=INT(RND(1)*13+1)
80 A2=INT(RND(1)*13+1) : B2=INT(RND(1)*13+1)
90 A3=INT(RND(1)*7)*2+1 : B3=INT(RND(1)*10)*2+1
100 A(K)=A1*A2*A3 : B(K)=B1*B2*B3
110 PRINT # 1,
120 PRINT # 1,"Num. ";K;TAB(9)"Halla el m.c.d. y el
m.c.m. de los números ";A(K);"y";B(K) : PRINT # 1,
130 NEXT K
140 REM Escritura de las soluciones
150 PRINT # 1, : PRINT # 1,
160 PRINT # 1, "***** S O L U C I O N E S *****"
170 FOR K=1 TO EJ
180 PRINT # 1, : PRINT # 1, "Num. ";K : PRINT # 1,
190 A=A(K) : B=B(K)
200 R=A MOD B : IF R<>0 THEN A=B : B=R : GOTO 200
210 M=A(K)*B(K)/B
220 PRINT # 1, "m.c.d. (";A(K);", ";B(K);") =";B
230 PRINT # 1, "m.c.m. (";A(K);", ";B(K);") =";M
240 NEXT K
245 CLOSE 1
250 END

```

En MSX la línea 47 queda así:

```

47 IF Z=1 THEN OPEN "LPT:" AS 1 ELSE OPEN "CRT:" AS 1

```

## 1.12 COMO PREPARAR UN DISCO AUTOEJECUTABLE

Usted puede preparar un disco para que lo utilicen aquellas personas que, estando interesadas en la G.A.E., no tienen muchos conocimientos sobre ordenadores.

Veamos los pasos que tiene que dar:

1. Coloque el disco del sistema operativo MS-DOS en *A* y un disco virgen en *B*.

2. Formatee el disco *B*, al tiempo que incluye el sistema operativo. Para ello teclee:

FORMAT B:/S

3. Copie el BASIC, que suponemos que está en el disco *A*. Para ello teclee:

COPY A:BASIC.COM B: (1)

4. Copie en el disco *B* los programas GAE. Para ello saque el disco del DOS de *A* e introduzca un disco que contenga los programas de este libro. Escriba:

COPY A:\*.BAS B:

5. Ahora puede trabajar con un solo disco. Introduzca en *A* el disco con las copias de los programas y modifique todos ellos cambiando la instrucción END por la SYSTEM.

6. Escriba los siguientes ficheros BATCH, procediendo así:

```
A>COPY CON: AUTOEXEC.BAT
echo off
indice
^Z
```

```
A>COPY CON: INDICE.BAT
echo off
cls
dir *.bas /w
echo Escriba :
echo      Ejecutar NOMBREDELPROGRAMA ( no necesita BAS )
^Z
```

```
A>COPY CON: EJECUTAR.BAT
echo off
basic %1
pause
indice
^Z
```

Si ha seguido los pasos anteriores, en el disco *A* debe tener los siguientes ficheros:

Sistema Operativo

BASIC

Programas GAE en la versión SYSTEM

AUTOEXEC.BAT

INDICE.BAT

EJECUTAR.BAT

Compruébelo escribiendo:

A>dir

Ahora pruebe el disco que acaba de preparar: ponga en marcha el ordenador con el disco *A* (apague y encienda, o bien pulse simultáneamente las teclas Ctrl Alt Del). En la pantalla aparecerá lo siguiente:

```
Volumen en unidad A sin etiqueta
Directorio de A:\
PRIMOS    BAS  MCDMCM    BAS  SIMPLIFI BAS  OPERAFRA BAS  CAMBSIST BAS
OPERSIST BAS  ECUACLIN BAS  ECUASEGU BAS  SUBROUTIN BAS  PRUEBA    BAS
PRUESUBR BAS  MULTIPOL BAS  DIVIPOLI BAS  DIVIRUFF BAS  VALORPOL BAS
FACTOPOL BAS  ECUACUAR BAS  FRACALGE BAS  DETERMIN BAS  SISTEMAS BAS
DISNORMA BAS  TESTSIMU BAS  DISTRIBI BAS  TRIANREC BAS  TRIANGUL BAS
GRAFICA   BAS

      26 Archivo(s)      193536 bytes libres

Escriba :
Ejecutar NOMBREDELPROGRAMA ( no necesita BAS )
A>
```

Escriba, por ejemplo,

Ejecutar mcdmcm                   (<RETURN>)

y el programa del máximo común divisor empezará a desarrollarse.



# 2

## DIVISIBILIDAD

### 2.1 DESCOMPOSICION DE UN NUMERO EN FACTORES PRIMOS

Con este programa el ordenador nos pregunta cuántos ejercicios queremos. A continuación nos propone la descomposición en factores de tantos números  $N$  como hayamos elegido y, por fin, nos muestra los resultados.

Veamos cómo funciona:

#### DESCOMPOSICION EN FACTORES PRIMOS

Cuántos ejercicios ? 10

Num. 1	Descomponer en factores primos el número	5460
Num. 2	Descomponer en factores primos el número	3857
Num. 3	Descomponer en factores primos el número	1386
Num. 4	Descomponer en factores primos el número	2156
Num. 5	Descomponer en factores primos el número	2730
Num. 6	Descomponer en factores primos el número	294
Num. 7	Descomponer en factores primos el número	165
Num. 8	Descomponer en factores primos el número	704
Num. 9	Descomponer en factores primos el número	8112
Num. 10	Descomponer en factores primos el número	357

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1		Num. 2		Num. 3		Num. 4		Num. 5	
5460	2	3857	7	1386	2	2156	2	2730	2
2730	2	551	19	693	3	1078	2	1365	3
1365	3	29	29	231	3	539	7	455	5
455	5	1		77	7	77	7	91	7
91	7			11	11	11	11	13	13
13	13			1		1		1	
1									

Num. 6		Num. 7		Num. 8		Num. 9		Num. 10	
294	2	165	3	704	2	8112	2	357	3
147	3	55	5	352	2	4056	2	119	7
49	7	11	11	176	2	2028	2	17	17
7	7	1		88	2	1014	2	1	
1				44	2	507	3		
				22	2	169	13		
				11	11	13	13		
				1		1			

Se ha procurado que el número  $N$  no sea excesivamente grande, que tenga variedad de factores primos, pero no superiores a 29, y que muy pocas veces sea primo. Esto se ha conseguido haciendo que  $N$  sea el producto de tres números aleatorios  $A1$ ,  $A2$  y  $A3$ ; de los cuales,  $A1$  y  $A2$  son números comprendidos entre 1 y 29, y  $A3$  es un número impar que varía entre 5 y 23.

He aquí el programa que produce las salidas anteriores:

```

10 REM Descomponer en factores primos
20 CLS
30 PRINT "DESCOMPOSICION EN FACTORES PRIMOS"
40 PRINT : PRINT
50 INPUT "Cuántos ejercicios ";EJ
60 DIM A(EJ)
70 FOR K=1 TO EJ
80 A1=INT(RND(1)*29+1)
90 A2=INT(RND(1)*29+1)
100 A3=INT(RND(1)*10)*2+5

```

```

110 A(K)=A1*A2*A3
120 PRINT
130 PRINT "Num.";K;"  Descomponer en factores primos
el número ";A(K)
140 NEXT K
150 REM Escritura de las soluciones
160 PRINT : PRINT
170 PRINT "***** S O L U C I O N E S *****"
180 FOR K=1 TO EJ
190 PRINT : PRINT "Num.";K : PRINT
200 N=A(K) : GOSUB 1000
210 NEXT K
220 END
1000 REM Subrutina primos
1010 L=LEN(STR$(N))
1020 D=1
1030 D=D+1
1040 IF D*D>N THEN GOTO 1090
1050 IF N MOD D<>0 THEN GOTO 1030
1060 PRINT TAB(L+1-LEN(STR$(N)))STR$(N);TAB(L+1)
CHR$(179);D
1070 N=N/D
1080 GOTO 1040
1090 PRINT TAB(L+1-LEN(STR$(N)))STR$(N);TAB(L+1)
CHR$(179);N
1100 PRINT TAB(L)"1";TAB(L+1);CHR$(179)
1110 RETURN

```

Si quiere que el programa vaya un poco más deprisa, incluya la línea

15 DEFINT A – Z

El carácter chr\$(179) de las líneas 1060, 1090 y 1100 es | (rayita vertical). En el listado lo puede sustituir por “|”.

Guarde el programa en el disco con el nombre de PRIMOS.BAS.

## 2.2 MAXIMO COMUN DIVISOR Y MINIMO COMUN MULTIPLO

Este programa propone tantos ejercicios como se deseen de cálculo de máximo común divisor y mínimo común múltiplo de dos números  $A$  y  $B$ . Al final da todas las soluciones.

Veamos una salida del programa:

MAXIMO COMUN DIVISOR Y MINIMO COMUN MULTIPLO

Cuántos ejercicios? 10

Num. 1 Halla el m.c.d. y el m.c.m. de los números 630 y 351

Num. 2 Halla el m.c.d. y el m.c.m. de los números 30 y 728

Num. 3 Halla el m.c.d. y el m.c.m. de los números 32 y 1428

Num. 4 Halla el m.c.d. y el m.c.m. de los números 7 y 280

Num. 5 Halla el m.c.d. y el m.c.m. de los números 66 y 1560

Num. 6 Halla el m.c.d. y el m.c.m. de los números 182 y 1360

Num. 7 Halla el m.c.d. y el m.c.m. de los números 45 y 147

Num. 8 Halla el m.c.d. y el m.c.m. de los números 864 y 108

Num. 9 Halla el m.c.d. y el m.c.m. de los números 9 y 55

Num. 10 Halla el m.c.d. y el m.c.m. de los números 792 y 180

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1

m.c.d.( 630 , 351 ) = 9  
m.c.m.( 630 , 351 ) = 24570

Num. 2

m.c.d.( 30 , 728 ) = 2  
m.c.m.( 30 , 728 ) = 10920

Num. 3

m.c.d.( 32 , 1428 ) = 4  
m.c.m.( 32 , 1428 ) = 11424

Num. 4

m.c.d.( 7 , 280 ) = 7  
m.c.m.( 7 , 280 ) = 280

Num. 5

m.c.d.( 66 , 1560 ) = 6  
m.c.m.( 66 , 1560 ) = 17160

Num. 6

m.c.d.( 182 , 1360 ) = 2  
m.c.m.( 182 , 1360 ) = 123760

Num. 7

m.c.d.( 45 , 147 ) = 3  
m.c.m.( 45 , 147 ) = 2205

Num. 8

m.c.d.( 864 , 108 ) = 108  
m.c.m.( 864 , 108 ) = 864

Num. 9

m.c.d.( 9 , 55 ) = 1  
m.c.m.( 9 , 55 ) = 495

Num. 10

m.c.d.( 792 , 180 ) = 36  
m.c.m.( 792 , 180 ) = 3960

El programa utiliza el algoritmo de Euclides para calcular el m.c.d.:

```
10 REM Máximo común divisor y mínimo común múltiplo
20 CLS
30 PRINT "MAXIMO COMUN DIVISOR Y MINIMO COMUN MULTI
PLO"
40 PRINT : PRINT : INPUT "Cuántos ejercicios";EJ
50 DIM A(EJ),B(EJ)
60 FOR K=1 TO EJ
70 A1=INT(RND(1)*13+1) : B1=INT(RND(1)*13+1)
80 A2=INT(RND(1)*13+1) : B2=INT(RND(1)*13+1)
90 A3=INT(RND(1)*7)*2+1 : B3=INT(RND(1)*10)*2+1
100 A(K)=A1*A2*A3 : B(K)=B1*B2*B3
110 PRINT
120 PRINT "Num. ";K;TAB(9) "Halla el m.c.d. y el
m.c.m. de los números ";A(K);"y";B(K) : PRINT
130 NEXT K
140 REM Escritura de las soluciones
150 PRINT : PRINT
```

```

160 PRINT "***** S O L U C I O N E S *****"
170 FOR K=1 TO EJ
180 PRINT : PRINT "Num.";K : PRINT
190 A=A(K) : B=B(K)
200 R=A MOD B : IF R<>0 THEN A=B : B=R : GOTO 200
210 M=A(K)*B(K)/B
220 PRINT "m.c.d. (";A(K);", ";B(K);") =";B
230 PRINT "m.c.m. (";A(K);", ";B(K);") =";M
240 NEXT K
250 END

```

Si quiere dar más rapidez al programa, incluya la línea

15 DEFSNG A-Z

Guárdelo en el disco con el nombre de MCDMCM.BAS.

# 3

## FRACCIONES

### 3.1 SIMPLIFICACION DE FRACCIONES

Con este programa el ordenador nos proporciona ejercicios de simplificación de fracciones y al final nos da las soluciones.

Veamos qué pasa cuando ejecutamos el programa:

SIMPLIFICACION DE FRACCIONES

Cuántos ejercicios ? 10

Num. 1      Simplifica la fracción :

$$\frac{3585}{3430} =$$

Num. 2      Simplifica la fracción :

$$\frac{1998}{1300} =$$

Num. 3      Simplifica la fracción :

$$\frac{375}{591} =$$

Num. 4      Simplifica la fracción :

$$\frac{143}{935} =$$

Num. 5      Simplifica la fracción :

$$\frac{681}{948} =$$

Num. 6      Simplifica la fracción :

$$\frac{4480}{2945} =$$

Num. 7      Simplifica la fracción :

$$\frac{13}{827} =$$

Num. 8      Simplifica la fracción :

$$\frac{1900}{2555} =$$

Num. 9      Simplifica la fracción :

$$\frac{976}{2936} =$$

Num. 10      Simplifica la fracción :

$$\frac{6118}{2933} =$$



\*\*\*\*\* SOLUCIONES \*\*\*\*\*

Num. 1

$$\frac{3585}{3430} = \frac{717}{686}$$

Num. 6

$$\frac{4480}{2945} = \frac{896}{589}$$

Num. 2

$$\frac{1998}{1300} = \frac{999}{650}$$

Num. 7

$$\frac{13}{827} = \frac{13}{827}$$

Num. 3

$$\frac{375}{591} = \frac{125}{197}$$

Num. 8

$$\frac{1900}{2555} = \frac{380}{511}$$

Num. 4

$$\frac{143}{935} = \frac{13}{85}$$

Num. 9

$$\frac{976}{2936} = \frac{122}{367}$$

Num. 5

$$\frac{681}{948} = \frac{227}{316}$$

Num. 10

$$\frac{6118}{2933} = \frac{874}{419}$$

En cada ejercicio, el ordenador genera al azar los dos términos de la fracción  $\frac{P(K)}{Q(K)}$ , como números comprendidos entre 10 y 999. Después calcula el máximo común divisor de ambos y los divide por él, para simplificar. Sin embargo, en la práctica resulta que la mayor parte de las veces, los números  $P(K)$  y  $Q(K)$  son primos entre sí. Si no se modifica el programa, apenas saldrán casos de simplificación.

Una posibilidad es rechazar esas parejas de números primos, hasta que resulte una que no lo sea, pero entonces la ejecución del programa resulta lenta.

En el programa se ha optado por aprovechar las parejas de números primos entre sí, para obtener otra fracción multiplicando ambos términos por un mismo número aleatorio comprendido entre 1 y 9, con lo cual se gana tiempo y se disminuye la probabilidad de que la fracción  $\frac{P(K)}{Q(K)}$  sea irreducible. Este cambio hace que aparezcan con cierta frecuencia números de cuatro cifras.

```

10 REM Simplificación de fracciones
20 CLS
30 RA$=STRING$(4,196)
40 PRINT "SIMPLIFICACION DE FRACCIONES"
50 PRINT : PRINT
60 INPUT "Cuántos ejercicios ";EJ
70 DIM P(EJ),Q(EJ),R(EJ),S(EJ)
80 DEF FNC$(X)=RIGHT$(" "+STR$(X),4)
90 FOR K=1 TO EJ
100 REM Generación de los terminos
110 P(K)=INT(RND(1)*990)+10
120 Q(K)=INT(RND(1)*990)+10
130 A=P(K) : B=Q(K) : GOSUB 350
140 IF B=1 THEN B=INT(RND(1)*9)+1 : P(K)=P(K)*B :
Q(K)=Q(K)*B
150 R(K)=P(K)/B : S(K)=Q(K)/B
160 REM Escritura
170 PRINT : PRINT : PRINT
180 PRINT "Num.";K; " Simplifica la fracción :"
190 PRINT

```

```

200 PRINT FNC$(P(K))
210 PRINT RA$+" = "
220 PRINT FNC$(Q(K))
230 NEXT K
240 REM Escritura de las soluciones
250 PRINT : PRINT
260 PRINT "***** SOLUCIONES *****"
270 FOR K=1 TO EJ
280 PRINT : PRINT
290 PRINT : PRINT "Num.";K : PRINT
300 PRINT FNC$(P(K));TAB(7) FNC$(R(K))
310 PRINT RA$+" = "+RA$
320 PRINT FNC$(Q(K));TAB(7) FNC$(S(K))
330 NEXT K
340 END
350 REM Subrutina m.c.d.
360 R=A MOD B : IF R<>0 THEN A=B : B=R : GOTO 360
370 RETURN

```

La línea 30 también se puede escribir así:

```
30 RA$="—"
```

Guarde este programa con el nombre de SIMPLIFI.BAS.

## 3.2 OPERACIONES CON FRACCIONES

Con este programa usted puede disponer de ejercicios de sumar, restar, multiplicar y dividir fracciones; todos ellos de una misma operación, o variados. Los resultados aparecen en la forma más simplificada.

Veamos qué ocurre cuando se ejecuta el programa:

### OPERACIONES CON FRACCIONES

Ejercicios de :

- 1 SUMAR
- 2 RESTAR
- 3 MULTIPLICAR
- 4 DIVIDIR
- 5 VARIADOS

Elija el número correspondiente ? 5

Cuántos ejercicios ? 10

Num. 1

$$\frac{16}{16} \times \frac{11}{22} =$$

Num. 6

$$\frac{20}{10} \times \frac{16}{2} =$$

Num. 2

$$\frac{3}{9} : \frac{13}{4} =$$

Num. 7

$$\frac{11}{18} : \frac{3}{13} =$$

Num. 3

$$\frac{5}{8} \times \frac{6}{20} =$$

Num. 8

$$\frac{19}{5} + \frac{12}{5} =$$

Num. 4

$$\frac{11}{2} - \frac{19}{2} =$$

Num. 9

$$\frac{7}{9} \times \frac{21}{6} =$$

Num. 5

$$\frac{12}{13} : \frac{3}{9} =$$

Num. 10

$$\frac{19}{16} : \frac{3}{2} =$$

\*\*\*\*\* SOLUCIONES \*\*\*\*\*

Num. 1

$$\frac{16}{16} \times \frac{11}{22} = \frac{1}{2}$$

Num. 6

$$\frac{20}{10} \times \frac{16}{2} = \frac{16}{1}$$

Num. 2

$$\frac{3}{9} : \frac{13}{4} = \frac{4}{39}$$

Num. 7

$$\frac{11}{18} : \frac{3}{13} = \frac{143}{54}$$

Num. 3

$$\frac{5}{8} \times \frac{6}{20} = \frac{3}{16}$$

Num. 8

$$\frac{19}{5} + \frac{12}{5} = \frac{31}{5}$$

Num. 4

$$\frac{11}{2} - \frac{19}{2} = \frac{-4}{1}$$

Num. 9

$$\frac{7}{9} \times \frac{21}{6} = \frac{49}{18}$$

Num. 5

$$\frac{12}{13} : \frac{3}{9} = \frac{36}{13}$$

Num. 10

$$\frac{19}{16} : \frac{3}{2} = \frac{19}{24}$$

He aquí el programa:

```
10 REM Operaciones con fracciones
20 CLS
30 PRINT "OPERACIONES CON FRACCIONES"
40 RA$=STRING$(2,196)
50 PRINT : PRINT
60 PRINT "  Ejercicios de :" : PRINT
70 PRINT "      1  SUMAR" : PRINT
80 PRINT "      2  RESTAR" : PRINT
90 PRINT "      3  MULTIPLICAR" : PRINT
100 PRINT "      4  DIVIDIR" : PRINT
110 PRINT "      5  VARIADOS" : PRINT
120 PRINT : PRINT
130 INPUT "Elija el número correspondiente ";N
140 H=N
150 PRINT : PRINT
160 INPUT "Cuántos ejercicios ";EJ
170 DIM P(EJ),Q(EJ),R(EJ),S(EJ),U(EJ),V(EJ),SI$(4),N(EJ)
180 SI$(1)=" + " : SI$(2)=" - " : SI$(3)=" x " : SI$(4)=" : "
190 DEF FNC$(X)=RIGHT$(STR$(X),2)
200 FOR K=1 TO EJ
210 REM Generación de los terminos
220 P(K)=INT(RND(1)*22+1)
230 Q(K)=INT(RND(1)*21+2)
240 R(K)=INT(RND(1)*22+1)
250 S(K)=INT(RND(1)*21+2)
260 REM Operación correspondiente
270 IF H=5 THEN N=INT(RND(1)*4+1) : N(K)=N
280 IF N=1 THEN U=P(K)*S(K)+R(K)*Q(K) : V=Q(K)*S(K) : GOTO 320
290 IF N=2 THEN U=P(K)*S(K)-R(K)*Q(K) : V=Q(K)*S(K) : GOTO 320
300 IF N=3 THEN U=P(K)*R(K) : V=Q(K)*S(K) : GOTO 320
310 IF N=4 THEN U=P(K)*S(K) : V=Q(K)*R(K)
320 REM Simplificación
330 A=U : B=V
```

```

340 R=A MOD B:IF R<>0 THEN A=B : B=R : GOTO 340
350 B=ABS(B) : U(K)=U/B : V(K)=V/B
360 REM Escritura del ejercicio
370 PRINT : PRINT : PRINT "Num. ";K : PRINT
380 PRINT FNC$(P(K))+ " " +FNC$(R(K))
390 PRINT RA$+SI$(N)+RA$+" ="
400 PRINT FNC$(Q(K))+ " " +FNC$(S(K))
410 NEXT K
420 REM Escritura de las soluciones
430 PRINT : PRINT
440 PRINT : PRINT : PRINT "***** SOLUCIONES *****"
450 FOR K=1 TO EJ
460 PRINT : PRINT : PRINT "Num. ";K : PRINT
470 IF H=5 THEN N=N(K)
480 PRINT FNC$(P(K))+ " " +FNC$(R(K))+ " " ;U(K)
490 PRINT RA$+SI$(N)+RA$+" = " +RA$+RA$
500 PRINT FNC$(Q(K))+ " " +FNC$(S(K))+ " " ;V(K)
510 NEXT K
520 END

```

La línea 40 también se puede escribir así:

```
40 RA$="—"
```

Guarde el programa con el nombre de OPERAFRA.BAS.





# 4

## SISTEMAS DE NUMERACION

### 4.1 CAMBIO DE BASE

Se presentan aquí los clásicos ejercicios de cambio de base en los sistemas de numeración: dado un número  $x$  escrito en base  $a$ , escribirlo en base  $b$ .

El ordenador elige, para cada ejercicio, la base inicial  $a$  y la base final  $b$ , de entre sendos intervalos  $[A1,A2]$  y  $[B1,B2]$ , previamente fijados por el usuario. Esto da una cierta flexibilidad al programa y, además, permite obtener todos los ejercicios con el mismo cambio de base, sin más que hacer  $A1=A2$  y  $B1=B2$ .

Veamos cómo funciona el programa:

#### SISTEMAS DE NUMERACION

##### Cambio de base

Bases de 2 a 16

base inicial entre A1 y A2

A1 , A2 ? 2,8

base final entre B1 y B2

B1 , B2 ? 5,10

Cuántos ejercicios ? 10

Num. 1

Pasar el número 10510, escrito en base 6 , a base 8

Num. 2

Pasar el número 2531, escrito en base 8 , a base 6

Num. 3

Pasar el número 5351, escrito en base 6 , a base 10

Num. 4

Pasar el número 131213, escrito en base 4 , a base 5

Num. 5

Pasar el número 4352, escrito en base 6 , a base 7

Num. 6

Pasar el número 4002, escrito en base 7 , a base 5

Num. 7

Pasar el número 13424, escrito en base 5 , a base 8

Num. 8

Pasar el número 131101, escrito en base 4 , a base 9

Num. 9

Pasar el número 33221, escrito en base 4 , a base 9

Num. 10

Pasar el número 12103, escrito en base 6 , a base 7

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1

10510 (en base 6 ) = 2712 (en base 8 )

Num. 2

2531 (en base 8 ) = 10201 (en base 6 )

Num. 3

5351 (en base 6 ) = 1219 (en base 10 )

Num. 4

131213 (en base 4 ) = 12435 (en base 6 )

Num. 5

4352 (en base 6 ) = 2633 (en base 7 )

Num. 6

4002 (en base 7 ) = 20444 (en base 5 )

Num. 7

13424 (en base 5 ) = 2132 (en base 8 )

Num. 8

131101 (en base 4 ) = 2511 (en base 9 )

Num. 9

33221 (en base 4 ) = 1332 (en base 9 )

Num. 10

12103 (en base 6 ) = 5103 (en base 7 )

Fijemos ahora la base inicial y la final:

## SISTEMAS DE NUMERACION

### Cambio de base

Bases de 2 a 16

base inicial entre A1 y A2

A1 , A2 ? 2,2

base final entre B1 y B2

B1 , B2 ? 10,10

Cuántos ejercicios ? 10

Num. 1

Pasar el número 10111001010, escrito en base 2 , a base 10

Num. 2

Pasar el número 10001101100, escrito en base 2 , a base 10

Num. 3

Pasar el número 10001101111, escrito en base 2 , a base 10

Num. 4

Pasar el número 10100011101, escrito en base 2 , a base 10

Num. 5

Pasar el número 11000110001, escrito en base 2 , a base 10

Num. 6

Pasar el número 11100100010, escrito en base 2 , a base 10

Num. 7

Pasar el número 10111100010, escrito en base 2 , a base 10

Num. 8

Pasar el número 10101010001, escrito en base 2 , a base 10

Num. 9

Pasar el número 10110000110, escrito en base 2 , a base 10

Num. 10

Pasar el número 11001000000, escrito en base 2 , a base 10

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1

10111001010 (en base 2 ) = 1482 (en base 10 )

Num. 2

10001101100 (en base 2 ) = 1132 (en base 10 )

Num. 3

10001101111 (en base 2 ) = 1135 (en base 10 )

Num. 4

10100011101 (en base 2 ) = 1309 (en base 10 )

Num. 5

11000110001 (en base 2 ) = 1585 (en base 10 )

Num. 6

11100100010 (en base 2 ) = 1826 (en base 10 )

Num. 7

10111100010 (en base 2 ) = 1506 (en base 10 )

Num. 8

10101010001 (en base 2 ) = 1361 (en base 10 )

Num. 9

10110000110 (en base 2 ) = 1414 (en base 10 )

Num. 10

11001000000 (en base 2 ) = 1600 (en base 10 )

Este es el programa de ejercicios de cambio de base en los sistemas de numeración:

```
10 REM Cambio de sistema de numeración
20 CLS
30 PRINT TAB(19)"SISTEMAS DE NUMERACION"
40 PRINT : PRINT
50 PRINT TAB(31)"Cambio de base"
60 PRINT : PRINT
70 PRINT "Bases de 2 a 16" : PRINT : PRINT
80 DIM CI$(15)
90 FOR I=0 TO 15 : READ CI$(I) : NEXT I
100 DATA 0,1,2,3,4,5,6,7,8,9,A,B,C,D,E,F
110 PRINT "base inicial entre A1 y A2" : PRINT
120 INPUT "A1 , A2 ";A1%,A2% : PRINT : PRINT
130 PRINT "base final entre B1 y B2" : PRINT
140 INPUT "B1 , B2 ";B1%,B2% : PRINT : PRINT
150 INPUT "Cuántos ejercicios ";EJ : PRINT
160 DIM A%(EJ),B%(EJ),X%(EJ)
170 FOR K=1 TO EJ
180 A%(K)=A1%+RND(1)*(A2%-A1%)
190 B%(K)=B1%+RND(1)*(B2%-B1%) : IF B%(K)=A%(K)
THEN GOTO 190
200 X%(K)=RND(1)*1000+1000
210 PRINT "Num.";K : PRINT
220 PRINT "Pasar el número ";
230 Z%=X%(K) : N%=A%(K) : GOSUB 390
240 PRINT B$;"", escrito en base ";A%(K); ", a base "
;B%(K)
250 PRINT : PRINT
260 NEXT K
270 REM Escritura de las soluciones
280 PRINT : PRINT
```

```

290 PRINT "***** S O L U C I O N E S *****"
300 FOR K=1 TO EJ
310 PRINT
320 PRINT "Num.";K : PRINT
330 Z%=X%(K) : N%=A%(K) : GOSUB 390
340 PRINT B$ ; " (en base";A%(K);") = ";
350 Z%=X%(K) : N%=B%(K) : GOSUB 390
360 PRINT B$;" (en base";B%(K);")"
370 NEXT K
380 END
390 REM Subrutina para pasar a base n
400 B$=""
410 IF Z%>=N% THEN B%=CI$(Z% MOD N%)+B$ : Z%=Z%\N%
: GOTO 410
420 B%=CI$(Z%)+B$
430 RETURN

```

Guarde el programa con el nombre de CAMBSIST.BAS.

## 4.2 SUMAR Y RESTAR EN DISTINTAS BASES

Con este programa se pueden obtener ejercicios de sumar y restar en una base determinada, junto con las soluciones. El programa permite elegir todos los ejercicios de la misma clase, por ejemplo, todos de sumar, o bien, variados: sumas y restas.

Funciona así:

### SISTEMAS DE NUMERACION

Elija la base ( hasta 16 ) ? 8

Ejercicios de :

- 1 SUMAR
- 2 RESTAR
- 3 VARIADOS

Elija el número correspondiente ? 3

Cuántos ejercicios ? 10

Num. 1 Restar en base 8

$$\begin{array}{r} 4476 \\ - 3654 \\ \hline \end{array}$$

Num. 6 Sumar en base 8

$$\begin{array}{r} 5134 \\ + 1757 \\ \hline \end{array}$$

Num. 2 Restar en base 8

$$\begin{array}{r} 4365 \\ - 2361 \\ \hline \end{array}$$

Num. 7 Sumar en base 8

$$\begin{array}{r} 3735 \\ + 3325 \\ \hline \end{array}$$

Num. 3 Sumar en base 8

$$\begin{array}{r} 4177 \\ + 2365 \\ \hline \end{array}$$

Num. 8 Restar en base 8

$$\begin{array}{r} 3273 \\ - 2314 \\ \hline \end{array}$$

Num. 4 Restar en base 8

$$\begin{array}{r} 3122 \\ - 2637 \\ \hline \end{array}$$

Num. 9 Restar en base 8

$$\begin{array}{r} 5272 \\ - 3444 \\ \hline \end{array}$$

Num. 5 Sumar en base 8

$$\begin{array}{r} 5346 \\ + 4172 \\ \hline \end{array}$$

Num. 10 Restar en base 8

$$\begin{array}{r} 4231 \\ - 1752 \\ \hline \end{array}$$

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Operaciones en base 8

Num. 1

$$\begin{array}{r} 4476 \\ - 3654 \\ \hline 622 \end{array}$$

Num. 2

$$\begin{array}{r} 4365 \\ - 2361 \\ \hline 2004 \end{array}$$



Num. 3

$$\begin{array}{r} 4177 \\ + 2365 \\ \hline 6564 \end{array}$$

Num. 7

$$\begin{array}{r} 3735 \\ + 3325 \\ \hline 7262 \end{array}$$

Num. 4

$$\begin{array}{r} 3122 \\ - 2637 \\ \hline 263 \end{array}$$

Num. 8

$$\begin{array}{r} 3273 \\ - 2314 \\ \hline 757 \end{array}$$

Num. 5

$$\begin{array}{r} 5346 \\ + 4172 \\ \hline 11540 \end{array}$$

Num. 9

$$\begin{array}{r} 5272 \\ - 3444 \\ \hline 1626 \end{array}$$

Num. 6

$$\begin{array}{r} 5134 \\ + 1757 \\ \hline 7113 \end{array}$$

Num. 10

$$\begin{array}{r} 4231 \\ - 1752 \\ \hline 2257 \end{array}$$

Este es el programa.

```
10 REM Sistemas de numeración. Sumas y restas
20 CLS
30 PRINT "SISTEMAS DE NUMERACION"
40 DIM CI$(15),SI$(2),OP$(2)
50 SI$(1)="+ " : SI$(2)="- "
60 OP$(1)="Sumar " : OP$(2)="Restar "
70 RA$=STRING$(15,196)
80 FOR I%=0 TO 15 : READ CI$(I%) : NEXT I%
90 DATA 0,1,2,3,4,5,6,7,8,9,A,B,C,D,E,F
100 PRINT : PRINT : PRINT
110 INPUT "Elija la base ( hasta 16 ) ";N%
120 PRINT : PRINT
130 PRINT "Ejercicios de :" : PRINT
```

```

140 PRINT "    1  SUMAR" : PRINT
150 PRINT "    2  RESTAR" : PRINT
160 PRINT "    3  VARIADOS" : PRINT : PRINT : PRINT
170 INPUT "Elija el número correspondiente ";M
180 H=M
190 PRINT : PRINT : INPUT "Cuántos ejercicios ";EJ
200 DIM X%(EJ),Y%(EJ),M(EJ)
210 FOR K=1 TO EJ
220 IF H=3 THEN M=INT(RND(1)*2+1) : M(K)=M
230 PRINT : PRINT
240 PRINT "Num.";K;OP$(M);"en base";N%
250 X%(K)=RND(1)*2000+1000
260 Y%(K)=RND(1)*2000+1000
270 IF X%(K)<Y%(K) THEN SWAP X%(K),Y%(K)
280 Z%=X%(K) : GOSUB 550
290 PRINT
300 PRINT TAB(15-LEN(B$))B$
310 Z%=Y%(K) : GOSUB 550
320 B$=SI$(M)+B$
330 PRINT TAB(15-LEN(B$))B$
340 PRINT TAB(15-LEN(B$))LEFT$(RA$,LEN(B$))
350 NEXT K
360 PRINT : PRINT
370 PRINT "***** S O L U C I O N E S *****"
380 PRINT : PRINT "Operaciones en base";N%
390 FOR K=1 TO EJ
400 IF H=3 THEN M=M(K)
410 PRINT
420 PRINT "Num.";K
430 PRINT
440 Z%=X%(K) : GOSUB 550
450 PRINT TAB(15-LEN(B$))B$
460 Z%=Y%(K) : GOSUB 550
470 B$=SI$(M)+B$
480 PRINT TAB(15-LEN(B$))B$
490 PRINT TAB(15-LEN(B$))LEFT$(RA$,LEN(B$))
500 IF M=1 THEN Z%=X%(K)+Y%(K) : GOSUB 550
510 IF M=2 THEN Z%=X%(K)-Y%(K) : GOSUB 550
520 PRINT TAB(15-LEN(B$))B$

```

```

530 NEXT K
540 END
550 REM Subrutina para pasar a base n
560 B$=""
570 IF Z%>=N% THEN B$=CI$(Z% MOD N%)+B$ : Z%=Z%\N%
: GOTO 570
580 B$=CI$(Z%)+B$
590 RETURN

```

La línea 60 se puede escribir así:

```
60 RA$="_____"
```

Guarde el programa con el nombre de OPERSIST.BAS.



# 5

## ECUACIONES

### 5.1. ECUACIONES LINEALES

Con este programa se obtienen ejercicios de ecuaciones lineales con denominadores, de la forma:

$$\frac{x+a}{f} - \frac{x-b}{g} + jx = -\frac{cx-d}{h} + \frac{e}{i}$$

Estas ecuaciones siempre tienen solución entera.

Así funciona el programa:

ECUACIONES LINEALES

Cuántos ejercicios? 10

Num. 1 Resuelve la ecuación :

$$\frac{x+7}{7} - \frac{x-5}{9} + 7x = -\frac{3x-4}{6} + \frac{-311}{6}$$

Num. 2 Resuelve la ecuación :

$$\frac{x+9}{3} - \frac{x-4}{3} + 9x = -\frac{6x-5}{2} + \frac{443}{6}$$

Num. 3 Resuelve la ecuación :

$$\frac{x+2}{4} - \frac{x-6}{6} + 2x = -\frac{4x-8}{8} + \frac{-14}{3}$$

Num. 4 Resuelve la ecuación :

$$\frac{x+7}{2} - \frac{x-6}{5} + 8x = -\frac{2x-6}{9} + \frac{331}{6}$$

Num. 5 Resuelve la ecuación :

$$\frac{x+3}{6} - \frac{x-3}{3} + 4x = -\frac{4x-9}{3} + \frac{11}{3}$$

Num. 6 Resuelve la ecuación :

$$\frac{x+8}{7} - \frac{x-3}{2} + 8x = -\frac{2x-4}{7} + \frac{-859}{14}$$

Num. 7 Resuelve la ecuación :

$$\frac{x+8}{2} - \frac{x-8}{4} + 3x = -\frac{2x-2}{2} + \frac{-65}{4}$$

Num. 8 Resuelve la ecuación :

$$\frac{x+4}{9} - \frac{x-3}{2} + 3x = -\frac{6x-6}{2} + \frac{-221}{18}$$

Num. 9 Resuelve la ecuación :

$$\frac{x+3}{2} - \frac{x-2}{6} + 3x = -\frac{5x-9}{2} + \frac{62}{3}$$

Num. 10 Resuelve la ecuación :

$$\frac{x+2}{6} - \frac{x-7}{3} + 8x = -\frac{6x-8}{9} + \frac{797}{18}$$

## Soluciones de ecuaciones lineales :

Num. 1	$x = -7$
Num. 2	$x = 6$
Num. 3	$x = -2$
Num. 4	$x = 6$
Num. 5	$x = 1$
Num. 6	$x = -8$
Num. 7	$x = -5$
Num. 8	$x = -2$
Num. 9	$x = 4$
Num. 10	$x = 5$

Este es el programa que proporciona los ejercicios de ecuaciones lineales:

```
10 REM Ecuaciones lineales
20 CLS
30 PRINT "ECUACIONES LINEALES" : PRINT : PRINT
40 R3$=STRING$(3,196) : R4$=STRING$(4,196)
50 INPUT "Cuántos ejercicios";EJ
60 DIM X(EJ),A(8),A$(8)
70 FOR K=1 TO EJ
80 REM Generación de las constantes
90 FOR I=1 TO 8
100 A(I)=INT(RND(1)*8+2)
110 A$(I)=RIGHT$(STR$(A(I)),1)
120 NEXT I
130 REM Generación de la solución
140 X(K)=INT(RND(1)*19-9)
150 REM Cálculo termino independiente
160 Y1=A(4)*A(8)*(X(K)+A(1))-A(2)*A(8)*(X(K)-A(3))+
A(2)*A(4)*A(8)*A(5)*X(K)+A(2)*A(4)*(A(6)*X(K)-A(7))
170 Y2=A(2)*A(4)*A(8)
180 A=Y1 : B=Y2
190 R=A-B*INT(A/B)
200 IF R<>0 THEN A=B : B=R : GOTO 190
210 Y1=Y1/B : IF ABS(Y1)>999 THEN GOTO 90
220 Y2=Y2/B : IF ABS(Y2)=1 THEN GOTO 90
230 REM Escritura de la ecuación
```

```

240 PRINT : PRINT : PRINT
250 PRINT "Num.";K;" Resuelve la ecuación :"
260 PRINT
270 PRINT"x+";A$(1);"    x-";A$(3);"                ";A$(6
);"x-";A$(7);"    ";Y1
280 PRINT R3$;" - ";R3$;" +";A$(5);"x = - ";R4$;
" + ";R4$
290 PRINT " ";A$(2);"    ";A$(4);"                ";
A$(8);"    ";Y2
300 NEXT K
310 REM Escritura de la solución
320 PRINT : PRINT : PRINT
330 PRINT"Soluciones de ecuaciones lineales :"
340 PRINT : PRINT
350 FOR K=1 TO EJ
360 PRINT "Num.";K;"    x = ";X(K)
370 NEXT K
380 END

```

La línea 30 se puede escribir así:

```
30 R3$+"____":R4$+"_____"
```

Grabe el programa en el disco con el nombre de ECUACLIN.BAS.

## 5.2 ECUACIONES DE SEGUNDO GRADO

El programa facilita la obtención de ejercicios de ecuaciones de segundo grado con soluciones enteras o fraccionarias.

Así funciona el programa:

ECUACIONES DE SEGUNDO GRADO

Cuántos ejercicios? 10

Num. 1 Resuelve la ecuación :

$$8X^2 - 22X + 12 = 0$$



Num. 2 Resuelve la ecuación :

$$6X^2 + 5X - 21 = 0$$

Num. 3 Resuelve la ecuación :

$$2X^2 + 6X - 56 = 0$$

Num. 4 Resuelve la ecuación :

$$6X^2 - 14X - 40 = 0$$

Num. 5 Resuelve la ecuación :

$$2X^2 + 12X - 54 = 0$$

Num. 6 Resuelve la ecuación :

$$2X^2 - X = 0$$

Num. 7 Resuelve la ecuación :

$$6X^2 - 32X + 42 = 0$$

Num. 8 Resuelve la ecuación :

$$8X^2 + 32X - 18 = 0$$

Num. 9 Resuelve la ecuación :

$$16X^2 + 28X - 8 = 0$$

Num. 10 Resuelve la ecuación :

$$X^2 + 5X - 6 = 0$$

Soluciones de la ecuación de 2° grado

Num. 1             $X_1 = 2$                              $X_2 = 3 / 4$

Num. 2             $X_1 = 3 / 2$                              $X_2 = -7 / 3$

Num. 3             $X_1 = -7$                                      $X_2 = 4$

Num. 4             $X_1 = -5 / 3$                              $X_2 = 4$

Num. 5             $X_1 = -9$                                      $X_2 = 3$

Num. 6             $X_1 = 0$                                      $X_2 = 1 / 2$

Num. 7             $X_1 = 3$                                      $X_2 = 7 / 3$

Num. 8             $X_1 = -9 / 2$                              $X_2 = 1 / 2$

Num. 9             $X_1 = -2$                                      $X_2 = 1 / 4$

Num. 10             $X_1 = -6$                                      $X_2 = 1$

El programa genera aleatoriamente los términos de las soluciones  $\frac{R_1}{S_1}, \frac{R_2}{S_2}$  y luego construye la ecuación correspondiente:

```

10 REM Ecuaciones de segundo grado
20 CLS
30 PRINT "ECUACIONES DE SEGUNDO GRADO": PRINT
40 DEFINT A-Z
50 INPUT "Cuántos ejercicios";EJ
60 DIM R1(EJ),R2(EJ),S1(EJ),S2(EJ)
70 FOR K=1 TO EJ
80 PRINT : PRINT
90 PRINT "Num.";K;" Resuelve la ecuación : " : PRINT
100 REM Generación de las soluciones
110 R1(K)=INT(RND(1)*19)-9 : S1(K)=INT(RND(1)*19)-9
120 R2(K)=INT(RND(1)*4)+1 : S2(K)=INT(RND(1)*4)+1
130 REM Construcción de la ecuación
140 A=R2(K)*S2(K)
150 B=-(S2(K)*R1(K)+S1(K)*R2(K))
160 C=R1(K)*S1(K)
170 IF A=1 THEN A$="X "
180 IF A=-1 THEN A$="-X "
190 IF A<-1 OR A>1 THEN A$=STR$(A)+"X "
200 IF B=0 THEN B$=""
210 IF B=1 THEN B$="+X"
220 IF B=-1 THEN B$="-X"
230 IF B<-1 THEN B$=STR$(B)+"X"
240 IF B>1 THEN B$="+"+RIGHT$(STR$(B),LEN(STR$(B))-1)+"X"
250 IF C=0 THEN C$=""
260 IF C<0 THEN C$=STR$(C)
270 IF C>0 THEN C$="+"+RIGHT$(STR$(C),LEN(STR$(C))-1)
280 EX$=SPACE$(LEN(A$)-1)+"2"
290 EC$=A$+B$+C$
300 REM Escritura de la ecuación
310 PRINT EX$
320 PRINT EC$+" = 0"
330 NEXT K
340 PRINT : PRINT : PRINT
350 REM Escritura de las soluciones
360 PRINT "Soluciones de la ecuación de 2° grado"
370 PRINT : PRINT

```

```

380 FOR K=1 TO EJ
390 A=R1(K) : B=R2(K) : GOSUB 530
400 R1=R1(K)/B : R2=R2(K)/B
410 A=S1(K) : B=S2(K) : GOSUB 530
420 S1=S1(K)/B : S2=S2(K)/B
430 R1$=STR$(R1) : S1$=STR$(S1)
440 IF R2=1 THEN R2$="" ELSE R2$=" /"+STR$(R2)
450 IF S2=1 THEN S2$="" ELSE S2$=" /"+STR$(S2)
460 IF R1=0 THEN R2$=""
470 IF S1=0 THEN S2$=""
480 PRINT"Num.";K;"      X1 =";R1$;R2$;TAB(30)"X2 ="
;S1$;S2$
490 PRINT : PRINT
500 NEXT K
510 END
520 REM Subrutina m.c.d.
530 R= A MOD B : IF R<>0 THEN  A=B : B=R : GOTO 530
540 B=ABS(B)
550 RETURN

```

Guarde este programa con el nombre de ECUASEGU.BAS.

# 6

## POLINOMIOS

### 6.1 ESCRITURA DE POLINOMIOS

Un polinomio queda determinado cuando conocemos su grado  $N$  y los coeficientes  $CP(N)$ ,  $CP(N-1)$ , ...,  $CP(1)$  y  $CP(0)$ , con el coeficiente del término de mayor grado  $CP(N)$  distinto de cero. Así, por ejemplo,  $N=4$  y  $CP(4) = -1$ ,  $CP(3) = 0$ ,  $CP(2) = 15$ ,  $CP(1) = -7$  y  $CP(0) = 31$ , nos da el polinomio:

$$-X^4 + 15X^2 - 7X + 31$$

La dificultad mayor que se presenta al trabajar con polinomios es la de su escritura en la pantalla o en el papel de la impresora, pues la impresión de uno de ellos exige dos líneas: una para los exponentes y otra para los coeficientes, los signos y las  $X$ . Líneas que hay que imprimir adecuadamente para que los exponentes caigan en su sitio.

El polinomio anterior lleva dos líneas de escritura:

línea de los exponentes :      4            2

línea de los coeficientes :  $-X + 15X - 7X + 31$

Obsérvese que cuando un coeficiente es 1 ó  $-1$ , no aparece el símbolo "1", que si el coeficiente es cero, no aparece el término correspondiente, y que el signo del coeficiente del término de mayor grado, cuando es positivo, no figura. Respecto de los exponentes hay que observar que cuando son 1 ó 0 no figuran, que tampoco figuran en el caso de que el coeficiente del término sea cero, y, en fin, que los espacios entre exponentes dependen del número de cifras de los coeficientes.

La subrutina que sigue prepara las cosas para poder escribir un polinomio en dos líneas con las sentencias:

PRINT EP\$ (escribe la línea de los exponentes)

PRINT CP\$ (escribe la línea de los coeficientes)

```
1000 REM Subrutina escritura polinomio
1010 BL$="      " : CP$="" : EP$=""
1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=""
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I)),
LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150
1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+"+RIGHT$(STR$(CP(0)),
LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>"" THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)
1180 FOR I=N TO 0 STEP -1
1190 CP$=CP$+CP$(I)
1200 NEXT I
1210 IF N=1 THEN RETURN
1220 FOR I=N TO 2 STEP -1
1230 IF CP(I)<>0 THEN EP$=EP$+LEFT$(BL$,LEN(CP$(
I))-2)+STR$(I)
1240 NEXT I
1250 RETURN
```

Las entradas de esta subrutina son, por un lado, el grado  $N$  del polinomio a escribir, y por otro, sus coeficientes  $CP(N)$ ,  $CP(N-1)$ , ...,  $CP(1)$ ,  $CP(0)$ . Estos valores deben ser proporcionados desde el programa principal. La subrutina devuelve el contenido de las variables de cadena

EP\$ (línea de los exponentes) y CP\$ (línea de los coeficientes, incluidos los signos y las X).

Desde el programa principal se escribirá el polinomio con las instrucciones:

PRINT EP\$:PRINT CP\$

Como la subrutina 1000 va a intervenir en varios de los programas que siguen, se aconseja vivamente al lector que la teclee y la compruebe inmediatamente con este pequeño programa:

```
10 REM Prueba de la subrutina
20 CLS
30 PRINT "PRUEBA DE LA SUBROUTINA" :PRINT : PRINT
40 INPUT "grado ";N
50 DIM CP(N),CP$(N)
60 FOR I=N TO 0 STEP -1
70 INPUT CP(I)
80 NEXT I
90 GOSUB 1000
100 PRINT : PRINT
110 PRINT EP$ : PRINT CP$
120 END
```

PRUEBA DE LA SUBROUTINA

```
grado ? 4
? -1
? 0
? 15
? -7
? 31
```

```
      4      2
- X +15X -7X+31
```

Ensaye la escritura de diversos polinomios y compruebe que la subrutina cumple todas las exigencias. Corrija los posibles errores y guárdela en el disco con

SAVE"SUBROUTIN.BAS",a

En adelante, cuando tenga que escribir un programa que lleve esta subrutina, cárguela previamente con

LOAD"SUBROUTIN.BAS"

y teclee el resto del programa.

Otra manera de proceder consiste en escribir primero el programa principal, y luego añadir la subrutina 1000 con

MERGE"SUBROUTIN.BAS"

## 6.2 MULTIPLICACION DE POLINOMIOS

Este programa propone tantos ejercicios como se necesiten de multiplicación de dos polinomios  $A(X)$  y  $B(X)$ . Al final proporciona una lista con todos los polinomios producto  $C(K)$ .

Los grados de  $A(X)$  y  $B(X)$  varían de un ejercicio a otro, pero siempre están comprendidos entre 2 y 4; por tanto, el grado de  $C(X)$  estará entre 4 y 8.

Veamos qué ocurre cuando ejecutamos el programa:

MULTIPLICACION DE POLINOMIOS

Cuántos ejercicios? 10

Num. 1 Multiplica los polinomios :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ -7X + 3X + 9X + 3 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ -5X + 8X - 7X + 2 \end{array}$$

Num. 2 Multiplica los polinomios :

$$\begin{array}{r} 2 \\ 2X + 8X - 5 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ -2X - 9X + 6X - 9 \end{array}$$



Num. 3 Multiplica los polinomios :

$$\begin{array}{r} 3 \quad 2 \\ 6X - 3X - 7X + 1 \end{array}$$

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ X + 2X - 9X + 4X - 2 \end{array}$$

Num. 4 Multiplica los polinomios :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ -6X + 6X + 7X + X - 8 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ -2X - 3X - 6X - 6 \end{array}$$

Num. 5 Multiplica los polinomios :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ -7X + 4X + 7X + X - 6 \end{array}$$

$$\begin{array}{r} 2 \\ -3X - 9X + 6 \end{array}$$

Num. 6 Multiplica los polinomios :

$$\begin{array}{r} 3 \quad 2 \\ 3X - 8X + X - 8 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ 4X + 9X + 4X + 5 \end{array}$$

Num. 7 Multiplica los polinomios :

$$\begin{array}{r} 2 \\ 3X + 4X + 1 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ -5X - 6X + 9X + 8 \end{array}$$

Num. 8 Multiplica los polinomios :

$$\begin{array}{r} 2 \\ 7X^2 - 9X + 5 \end{array}$$

$$\begin{array}{r} 2 \\ -9X^2 - 8X - 6 \end{array}$$

Num. 9 Multiplica los polinomios :

$$\begin{array}{r} 2 \\ -4X^2 - 8X - 5 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ -8X^3 - 6X^2 + 7X - 8 \end{array}$$

Num. 10 Multiplica los polinomios :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ -9X^4 - 6X^3 + 9X^2 - 3X - 5 \end{array}$$

$$\begin{array}{r} 2 \\ -9X^2 + 2X + 2 \end{array}$$

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1

$$\begin{array}{r} 7 \quad 6 \quad 5 \quad 4 \quad 3 \quad 2 \\ 35X^7 - 71X^6 + 28X^5 + 37X^4 - 72X^3 + 42X^2 - 21X + 6 \end{array}$$

Num. 2

$$\begin{array}{r} 5 \quad 4 \quad 3 \quad 2 \\ -4X^5 - 34X^4 - 50X^3 + 75X^2 - 102X + 45 \end{array}$$

Num. 3

$$\begin{array}{r} 7 \quad 6 \quad 5 \quad 4 \quad 3 \quad 2 \\ 6X^7 + 9X^6 - 67X^5 + 38X^4 + 41X^3 - 31X^2 + 18X - 2 \end{array}$$

Num. 4

$$12X^7 + 6X^6 + 4X^5 - 23X^4 - 65X^3 - 24X^2 + 42X + 48$$

Num. 5

$$21X^6 + 51X^5 - 99X^4 - 42X^3 + 51X^2 + 60X - 36$$

Num. 6

$$12X^6 - 5X^5 - 56X^4 - 40X^3 - 108X^2 - 27X - 40$$

Num. 7

$$-15X^5 - 38X^4 - 2X^3 + 54X^2 + 41X + 8$$

Num. 8

$$-63X^4 + 25X^3 - 15X^2 + 14X - 30$$

Num. 9

$$32X^5 + 88X^4 + 60X^3 + 6X^2 + 29X + 40$$

Num. 10

$$81X^6 + 36X^5 - 111X^4 + 33X^3 + 57X^2 - 16X - 10$$

He aquí el programa:

```
10 REM Multiplicación de polinomios
20 CLS
30 PRINT "MULTIPLICACION DE POLINOMIOS"
40 PRINT : PRINT
50 DEFINT A-Z
60 INPUT "Cuántos ejercicios";EJ
70 DIM A(8),B(8),CP(8),CP$(8),G(EJ),C(EJ,8)
80 FOR K=1 TO EJ
90 PRINT : PRINT
100 PRINT "Num.";K;"Multiplica los polinomios : "
110 FOR I=0 TO 4 : A(I)=0 : B(I)=0 : NEXT I
120 GA =INT(RND(1)*3)+2
130 FOR I=0 TO GA
140 A(I)=INT(RND(1)*19)-9
150 NEXT I
160 IF A(GA)=0 THEN A(GA)=1
170 GB=INT(RND(1)*3)+2
180 FOR I=0 TO GB
190 B(I)=INT(RND(1)*19)-9
200 NEXT I
210 IF B(GB)=0 THEN B(GB)=1
220 FOR I=0 TO GA : CP(I)=A(I) : NEXT I
230 N=GA
240 GOSUB 1000
250 PRINT : PRINT EP$ : PRINT CP$
260 FOR I=0 TO GB : CP(I)=B(I) : NEXT I
270 N=GB
280 GOSUB 1000
290 PRINT : PRINT EP$ : PRINT CP$
300 REM Multiplicación
310 G(K)=GA+GB
320 FOR I=0 TO G(K)
330 FOR J=0 TO I
340 C(K,I)=C(K,I)+A(J)*B(I-J)
350 NEXT J
360 NEXT I
370 NEXT K
380 REM Escritura de las soluciones
390 PRINT : PRINT
```

```

400 PRINT "***** S O L U C I O N E S *****"
410 FOR K=1 TO EJ
420 PRINT : PRINT : PRINT "Num. ";K
430 N=G(K) : FOR I=0 TO G(K) : CP(I)=C(K,I) : NEXT I
440 GOSUB 1000
450 PRINT : PRINT : PRINT EP$ : PRINT CP$
460 NEXT K
470 END

1000 REM Subrutina escritura polinomio
1010 BL$=" " : CP$="" : EP$=""
1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=""
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I)),
LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150
1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+"+RIGHT$(STR$(CP(0)),
LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>"" THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)
1180 FOR I=N TO 0 STEP -1
1190 CP$=CP$+CP$(I)
1200 NEXT I
1210 IF N=1 THEN RETURN
1220 FOR I=N TO 2 STEP -1
1230 IF CP(I)<>0 THEN EP$=EP$+LEFT$(BL$,LEN(CP$(
I))-2)+STR$(I)
1240 NEXT I
1250 RETURN

```

Compruebe el programa y guárdelo en el disco con el nombre de MULTIPOL.BAS.

### 6.3 DIVISION DE POLINOMIOS

Este programa no divide los polinomios, sino que prepara ejercicios de división. Para ello elige aleatoriamente el divisor, el cociente y el resto, y calcula el dividendo. A continuación escribe el dividendo junto con el divisor, y reserva el cociente y el resto para mostrarlos con las soluciones.

Veamos qué pasa cuando ejecutamos el programa:

#### DIVISION DE POLINOMIOS

Cuántos ejercicios? 10

Num. 1 Divide los polinomios :

$$\begin{array}{r} 5 \quad 4 \quad 3 \quad 2 \\ -21X^5 - 57X^4 + 12X^3 - 44X^2 + 2X - 11 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ 3X^3 + 9X^2 + 3 \end{array}$$

Num. 2 Divide los polinomios :

$$\begin{array}{r} 5 \quad 4 \quad 3 \quad 2 \\ 6X^5 - 9X^4 - 13X^3 + 20X^2 + 2X - 7 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ 6X^3 - 9X^2 - X + 2 \end{array}$$

Num. 3 Divide los polinomios :

$$\begin{array}{r} 6 \quad 5 \quad 4 \quad 3 \quad 2 \\ -9X^6 + 76X^5 - 79X^4 + 43X^3 - 65X^2 + 29X + 6 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ -9X^3 + 4X^2 - 2X + 7 \end{array}$$

Num. 4 Divide los polinomios :

$$\begin{array}{r} 5 \quad 4 \quad 3 \quad 2 \\ -8X^5 - 26X^4 - 47X^3 - 33X^2 + 12X + 27 \end{array}$$

$$\begin{array}{r} 2 \\ -2X^2 - 3X - 6 \end{array}$$

Num. 5 Divide los polinomios :

$$\begin{array}{r} 6 \quad 5 \quad 4 \quad 3 \quad 2 \\ -40X^6 + 23X^5 - 42X^4 + 31X^3 - 29X^2 + X + 81 \end{array}$$

$$\begin{array}{r} 3 \quad 2 \\ -8X^3 + 3X^2 - 3X - 9 \end{array}$$

Num. 6 Divide los polinomios :

$$\begin{array}{r} 6 \quad 5 \quad 4 \quad 3 \quad 2 \\ 40X^6 - 72X^5 - 25X^4 - 55X^3 - 51X^2 - 38X - 17 \end{array}$$

$$\begin{array}{r} 3 \\ 8X^3 + 3X + 4 \end{array}$$

Num. 7 Divide los polinomios :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ 36X^4 + 104X^3 + 133X^2 + 95X + 22 \end{array}$$

$$\begin{array}{r} 2 \\ -9X^2 - 8X - 6 \end{array}$$

Num. 8 Divide los polinomios :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ 45X^4 - 6X^3 + 3X^2 + 84X - 70 \end{array}$$

$$\begin{array}{r} 2 \\ -5X^2 + 4X - 8 \end{array}$$

Num. 9 Divide los polinomios :

$$\begin{array}{r} 5 \quad 4 \quad 3 \quad 2 \\ 9X^5 + 68X^4 + 37X^3 + 14X^2 + 67X - 8 \\ \\ 3 \quad 2 \\ -9X^3 - 5X^2 - 2X - 9 \end{array}$$

Num. 10 Divide los polinomios :

$$\begin{array}{r} 6 \quad 5 \quad 4 \quad 3 \quad 2 \\ -7X^6 - 28X^5 + 75X^4 - 22X^3 + 41X^2 - 30X + 18 \\ \\ 3 \quad 2 \\ X^3 + 5X^2 - 5X + 3 \end{array}$$

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1

$$\begin{array}{r} 2 \\ \text{Cociente : } -7X^2 + 2X - 2 \\ \\ 2 \\ \text{Resto : } -5X^2 - 4X - 5 \end{array}$$

Num. 2

$$\begin{array}{r} 2 \\ \text{Cociente : } X^2 - 2 \\ \\ \text{Resto : } -3 \end{array}$$

Num. 3

$$\begin{array}{r} 3 \quad 2 \\ \text{Cociente : } X^3 - 8X^2 + 5X \\ \\ 2 \\ \text{Resto : } X^2 - 6X + 6 \end{array}$$

Num. 4

$$\begin{array}{r} 3 \quad 2 \\ \text{Cociente : } 4X^3 + 7X^2 + X - 6 \\ \\ \text{Resto : } -9 \end{array}$$



Num. 5

$$\text{Cociente : } 5X^3 - X^2 + 3X - 8$$

$$\text{Resto : } -5X + 4X + 9$$

Num. 6

$$\text{Cociente : } 5X^3 - 9X^2 - 5X - 6$$

$$\text{Resto : } 7$$

Num. 7

$$\text{Cociente : } -4X^2 - 8X - 5$$

$$\text{Resto : } 7X - 8$$

Num. 8

$$\text{Cociente : } -9X^2 - 6X + 9$$

$$\text{Resto : } 2$$

Num. 9

$$\text{Cociente : } -X^2 - 7X$$

$$\text{Resto : } -9X^2 + 4X - 8$$

Num. 10

$$\text{Cociente : } -7X^3 + 7X^2 + 5X + 9$$

$$\text{Resto : } -9$$

Este es el programa:

```
10 REM División de polinomios
20 CLS
30 PRINT "DIVISION DE POLINOMIOS"
40 DEFINT A-Z
```

```

50 PRINT : PRINT
60 INPUT "Cuántos ejercicios";EJ
70 DIM D(6),DI(6),GC(EJ),C(EJ,6),GR(EJ),R(EJ,6)
,CP(6),CP$(6)
80 FOR K=1 TO EJ
90 FOR I=0 TO 6 : D(I)=0 : DI(I)=0 : NEXT I
100 PRINT : PRINT
110 PRINT "Num.";K;"Divide los polinomios :"
120 PRINT
130 GD=INT(RND(1)*2)+2
140 FOR I=0 TO GD
150 D(I)=INT(RND(1)*19)-9
160 NEXT I
170 IF D(GD)=0 THEN D(GD)=1
180 GC(K)=INT(RND(1)*2)+2
190 FOR I=0 TO GC(K)
200 C(K,I)=INT(RND(1)*19)-9
210 NEXT I
220 IF C(K,GC(K))=0 THEN C(K,GC(K))=1
230 GR(K)=INT(RND(1)*GD)
240 FOR I=0 TO GR(K)
250 R(K,I)=INT(RND(1)*19)-9
260 NEXT I
270 IF R(K,GR(K))=0 THEN R(K,GR(K))=1
280 REM Cálculo del dividendo
290 G=GD+GC(K)
300 FOR I=0 TO G
310 DI(I)=0
320 FOR J=0 TO I
330 DI(I)=DI(I)+D(J)*C(K,I-J)
340 NEXT J
350 NEXT I
360 N=G : FOR I=0 TO N:CP(I)=DI(I)+R(K,I):NEXT I
370 GOSUB 1000
380 PRINT EP$ : PRINT CP$
390 N=GD : FOR I=0 TO N : CP(I)=D(I) : NEXT I
400 GOSUB 1000
410 PRINT : PRINT EP$ : PRINT CP$
420 NEXT K

```

```

430 REM Escritura de las soluciones
440 PRINT : PRINT
450 PRINT "***** S O L U C I O N E S *****"
460 FOR K=1 TO EJ
470 PRINT : PRINT : PRINT "Num.";K
480 N=GC(K)
490 FOR I=0 TO N : CP(I)=C(K,I) : NEXT I
500 GOSUB 1000
510 PRINT TAB(11) EP$
520 PRINT "Cociente :";TAB(11) CP$
530 N= GR(K)
540 FOR I=0 TO N : CP(I)=R(K,I) : NEXT I
550 GOSUB 1000
560 PRINT TAB(11) EP$
570 PRINT "Resto :";TAB(11) CP$
580 NEXT K
590 END
1000 REM Subrutina escritura polinomio
1010 BL$=" " : CP$="" : EP$=""
1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=" "
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I))
,LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150
1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+"+RIGHT$(STR$(CP(0))
,LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>" " THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)

```

```

1180 FOR I=N TO 0 STEP -1
1190 CP$=CP$+CP$(I)
1200 NEXT I
1210 IF N=1 THEN RETURN
1220 FOR I=N TO 2 STEP -1
1230 IF CP(I)<>0 THEN EP$=EP$+LEFT$(BL$,LEN(CP$(
I))-2)+STR$(I)
1240 NEXT I
1250 RETURN

```

Pruebe el programa y grábelo en el disco con el nombre de DIVIPOLI.BAS.

## 6.4 DIVISION DE RUFFINI

Este programa propone ejercicios de dividir un polinomio, de grado comprendido ente 2 y 5, por un polinomio de la forma  $x - a$ . En este caso el ordenador sí hace la división. Para ello utiliza el algoritmo de Ruffini y conserva, en cada ejercicio, el cociente y el resto, para escribirlos entre las soluciones.

Cuando se ejecuta el programa sale:

### DIVISION POR RUFFINI

Cuántos ejercicios? 10

Num. 1

Divide el polinomio :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ -6X^4 + 2X^3 + 7X^2 + 3 \end{array} \quad \text{por} \quad x+1$$

Num. 2

Divide el polinomio :

$$\begin{array}{r} 4 \quad 3 \quad 2 \\ -4X^4 - 3X^3 - 4X^2 + 7X - 5 \end{array} \quad \text{por} \quad x-4$$

Num. 3

Divide el polinomio :

$$-2X^4 - 7X^3 + 5X^2 - 7X - 1 \quad \text{por} \quad x+1$$

Num. 4

Divide el polinomio :

$$-X^4 + 6X^3 + 5X^2 - 2X - 6 \quad \text{por} \quad x-2$$

Num. 5

Divide el polinomio :

$$4X^2 + 2 \quad \text{por} \quad x+4$$

Num. 6

Divide el polinomio :

$$-5X^4 - 5X^2 + 5X + 6 \quad \text{por} \quad x+3$$

Num. 7

Divide el polinomio :

$$X^3 - 5X^2 + 6X - 2 \quad \text{por} \quad x-4$$

Num. 8

Divide el polinomio :

$$-3X^4 - 7X^3 + 5X^2 - 7X - 6 \quad \text{por} \quad x-2$$

Num. 9

Divide el polinomio :

$$2x^2 - 6x + 1 \quad \text{por} \quad x + 1$$

Num. 10

Divide el polinomio :

$$x^4 - 4x^3 + 3x^2 + 7x + 3 \quad \text{por} \quad x - 2$$

\*\*\*\*\* SOLUCIONES \*\*\*\*\*

Num. 1

Cociente :

$$-6x^3 + 8x^2 - x + 1$$

$$\text{Resto} = 2$$

Num. 2

Cociente :

$$-4x^3 - 19x^2 - 80x - 313$$

$$\text{Resto} = -1257$$

Num. 3

Cociente :

$$-2x^3 - 5x^2 + 10x - 17$$

$$\text{Resto} = 16$$

Num. 4

Cociente :

$$-x^3 + 4x^2 + 13x + 24$$

$$\text{Resto} = 42$$

Num. 5  
Cociente :

$$4X-16$$

$$\text{Resto} = 66$$

Num. 6  
Cociente :

$$\begin{array}{r} 3 \quad 2 \\ -5X + 15X - 50X + 155 \end{array}$$

$$\text{Resto} = -459$$

Num. 7  
Cociente :

$$\begin{array}{r} 2 \\ X - X + 2 \end{array}$$

$$\text{Resto} = 6$$

Num. 8  
Cociente :

$$\begin{array}{r} 3 \quad 2 \\ -3X - 13X - 21X - 49 \end{array}$$

$$\text{Resto} = -104$$

Num. 9  
Cociente :

$$2X-8$$

$$\text{Resto} = 9$$

Num. 10  
Cociente :

$$\begin{array}{r} 3 \quad 2 \\ X - 2X - X + 5 \end{array}$$

$$\text{Resto} = 13$$

He aquí el programa:

```
10 REM División Ruffini
20 CLS
30 PRINT "DIVISION POR RUFFINI" : PRINT : PRINT
40 INPUT "Cuántos ejercicios";EJ
50 DIM C(EJ,5),R(EJ),N(EJ),CP(5),CC(5)
60 FOR K=1 TO EJ
70 PRINT : PRINT : PRINT"Num.";K
80 N=INT(RND(1)*4+2)
90 FOR J=0 TO N
100 CP(J)=INT(RND(1)*15-7)
110 NEXT J
120 IF CP(N)=0 THEN CP(N)=2
130 A=INT(RND(1)*11)-5 : IF A=0 THEN A=-1
140 GOSUB 1000
150 PRINT : PRINT "Divide el polinomio :" : PRINT
160 PRINT EP$ : PRINT CP$;
170 PRINT "      por      x";
180 IF A>0 THEN PRINT -A ELSE PRINT "+";RIGHT$
   (STR$(A),1)
190 GOSUB 2000
200 N(K)=N-1 : FOR I=0 TO N(K):C(K,I)=CC(I):NEXT I
210 R(K)=R
220 NEXT K
230 REM Escritura de las soluciones
240 PRINT : PRINT : PRINT "***** SOLUCIONES *****"
250 PRINT : PRINT
260 FOR K=1 TO EJ
270 PRINT "Num.";K
280 N=N(K)
290 FOR I=0 TO N:CP(I)=C(K,I):NEXT I : GOSUB 1000
300 PRINT "Cociente :" : PRINT
310 PRINT EP$
320 PRINT CP$ : PRINT
330 PRINT "Resto =" : R(K) : PRINT
340 NEXT K
350 END
1000 REM Subrutina escritura polinomio
```



```

1010 BL$="      " : CP$="" : EP$=""
1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=""
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I)),
,LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150
1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+"+RIGHT$(STR$(CP(0)),
,LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>"" THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)
1180 FOR I=N TO 0 STEP -1
1190 CP$=CP$+CP$(I)
1200 NEXT I
1210 IF N=1 THEN RETURN
1220 FOR I=N TO 2 STEP -1
1230 IF CP(I)<>0 THEN EP$=EP$+LEFT$(BL$,LEN(CP$(
I))-2)+STR$(I)
1240 NEXT I
1250 RETURN
2000 REM Subrutina Ruffini
2010 CC(N-1)=CP(N)
2020 FOR I=N-1 TO 1 STEP -1
2030 CC(I-1)=CC(I)*A+CP(I)
2040 NEXT I
2050 R=CC(0)*A+CP(0)
2060 RETURN

```

Teclee el programa y guárdelo con el nombre de DIVIRUFF.BAS.

## 6.5 VALOR NUMERICO DE UN POLINOMIO

Con este programa el ordenador elige al azar un polinomio  $P(X)$  y un número  $a$ , y con el algoritmo de Ruffini calcula  $P(a)$ . Este valor lo reserva para mostrarlo en las soluciones.

Veamos qué ocurre cuando se pone en marcha el programa:

### VALOR NUMERICO DE UN POLINOMIO

Cuántos ejercicios? 10

Num. 1 Calcula el valor numérico del polinomio :

$$-6X^4 + 2X^3 + 7X^2 + 3 \quad \text{para } X = -2$$

Num. 2 Calcula el valor numérico del polinomio :

$$-4X^4 - 3X^3 - 4X^2 + 7X - 5 \quad \text{para } X = 6$$

Num. 3 Calcula el valor numérico del polinomio :

$$-2X^4 - 7X^3 + 5X^2 - 7X - 1 \quad \text{para } X = 2$$

Num. 4 Calcula el valor numérico del polinomio :

$$-X^4 + 6X^3 + 5X^2 - 2X - 6 \quad \text{para } X = 3$$

Num. 5 Calcula el valor numérico del polinomio :

$$4X^2 + 2 \quad \text{para } X = -6$$

Num. 6 Calcula el valor numérico del polinomio :

$$-5X^4 - 5X^3 + 5X^2 + 6 \quad \text{para } X = -5$$

Num. 7 Calcula el valor numérico del polinomio :

$$X^3 - 5X^2 + 6X - 2 \quad \text{para } X = 5$$

Num. 8 Calcula el valor numérico del polinomio :

$$-3X^4 - 7X^3 + 5X^2 - 7X - 6 \quad \text{para } X = 2$$

Num. 9 Calcula el valor numérico del polinomio :

$$2X^2 - 6X + 1 \quad \text{para } X = -1$$

Num. 10 Calcula el valor numérico del polinomio :

$$X^4 - 4X^3 + 3X^2 + 7X + 3 \quad \text{para } X = 3$$

\*\*\*\*\* SOLUCIONES \*\*\*\*\*

Num. 1     -81                      Num. 6     -3269

Num. 2     -5939                   Num. 7        28

Num. 3     -83                      Num. 8     -104

Num. 4     -114                     Num. 9        9

Num. 5       146                     Num. 10     24

Este es el programa:

```
10 REM Valor numérico de un polinomio
20 CLS
30 PRINT "VALOR NUMERICO DE UN POLINOMIO"
40 PRINT : PRINT
50 INPUT "Cuántos ejercicios";EJ
```

```

60 DIM R(EJ),CP(5),CC(4)
70 FOR K=1 TO EJ
80 N=INT(RND(1)*4+2)
90 FOR J=0 TO N
100 CP(J)=INT(RND(1)*15-7)
110 NEXT J
120 IF CP(N)=0 THEN CP(N)=1
130 A=INT(RND(1)*15)-7 : IF A=0 THEN A=2
140 GOSUB 1000
150 PRINT
160 PRINT "Num.":K;"Calcula el valor numérico del
polinomio : "
170 PRINT
180 PRINT EP$ : PRINT CP$;"      para X=":A
190 GOSUB 2000
200 R(K)=R
210 NEXT K
220 REM Escritura de las soluciones
230 PRINT : PRINT : PRINT "***** SOLUCIONES *****"
240 FOR K=1 TO EJ
250 PRINT : PRINT : PRINT "Num.":K; TAB(10) R(K)
260 NEXT K
270 END
1000 REM Subrutina escritura polinomio
1010 BL$="      " : CP$="" : EP$=""
1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=""
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I)),
LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150

```

```

1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+" +RIGHT$(STR$(CP(0))
,LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>" " THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)
1180 FOR I=N TO 0 STEP -1
1190 CP$=CP$+CP$(I)
1200 NEXT I
1210 IF N=1 THEN RETURN
1220 FOR I=N TO 2 STEP -1
1230 IF CP(I)<>0 THEN EP$=EP$+LEFT$(BL$,LEN(CP$(
I))-2)+STR$(I)
1240 NEXT I
1250 RETURN
2000 REM Subrutina Ruffini
2010 CC(N-1)=CP(N)
2020 FOR I=N-1 TO 1 STEP -1
2030 CC(I-1)=CC(I)*A+CP(I)
2040 NEXT I
2050 R=CC(0)*A+CP(0)
2060 RETURN

```

Guarde este programa con el nombre de VALORPOL.BAS.

## 6.6 FACTORIZACION DE POLINOMIOS

El programa saca al azar cuatro números enteros, comprendidos entre -5 y 5, y prepara un polinomio que los tenga como raíces. Al final muestra la descomposición en factores de todos los polinomios.

Vamos:

FACTORIZACION DE POLINOMIOS

Cuántos ejercicios? 10

Num. 1

Descomponer en factores el polinomio :

$$X^4 - 11X^3 + 42X^2 - 68X + 40$$

Num. 2

Descomponer en factores el polinomio :

$$-2X^4 + 2X^3 + 42X^2 - 2X - 40$$

Num. 3

Descomponer en factores el polinomio :

$$-X^4 + 15X^2 + 10X - 24$$

Num. 4

Descomponer en factores el polinomio :

$$-X^4 - 7X^3 + 9X^2 + 115X + 100$$

Num. 5

Descomponer en factores el polinomio :

$$-X^4 - X^3 + 18X^2 + 16X - 32$$

Num. 6

Descomponer en factores el polinomio :

$$2X^4 + 6X^3 - 22X^2 - 6X + 20$$

Num. 7

Descomponer en factores el polinomio :

$$-X^4 + 5X^3 + 10X^2 - 80X + 96$$

Num. 8

Descomponer en factores el polinomio :

$$2X^4 + 18X^3 + 36X^2 - 64X - 192$$

Num. 9

Descomponer en factores el polinomio :

$$-X^4 + 2X^3 + 16X^2 - 2X - 15$$

Num. 10

Descomponer en factores el polinomio :

$$2X^4 + 6X^3 - 52X^2 - 96X + 320$$

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

- Num. 1             $(X-2)(X-2)(X-5)(X-2)$   
Num. 2             $-2(X+1)(X-1)(X+4)(X-5)$   
Num. 3             $-(X+2)(X+3)(X-4)(X-1)$   
Num. 4             $-(X+5)(X-4)(X+5)(X+1)$   
Num. 5             $-(X-1)(X+4)(X+2)(X-4)$   
Num. 6             $2(X+1)(X-2)(X+5)(X-1)$   
Num. 7             $-(X-3)(X+4)(X-2)(X-4)$   
Num. 8             $2(X+4)(X-2)(X+4)(X+3)$   
Num. 9             $-(X+1)(X-5)(X+3)(X-1)$   
Num. 10            $2(X-2)(X+4)(X+5)(X-4)$

Este es el programa que lo realiza:

```
10 REM Factorizar polinomios
20 CLS
30 PRINT "FACTORIZACION DE POLINOMIOS" : PRINT
40 N=4
50 INPUT "Cuántos ejercicios";EJ
60 DIM R(4),RA(EJ,4),A(EJ)
```

```

70 FOR K=1 TO EJ
80 A=INT(RND(1)*5)-2 : IF A=0 THEN A=-1
90 FOR I=1 TO 4
100 R(I)=INT(RND(1)*11)-5 : IF R(I)=0 THEN R(I)=2
110 NEXT I
120 CP(4)=A
130 CP(3)=-(R(1)+R(2)+R(3)+R(4))*A
140 CP(2)=(R(1)*R(2)+R(1)*R(3)+R(1)*R(4)+R(2)*R(3)
)+R(2)*R(4)+R(3)*R(4))*A
150 CP(1)=-(R(1)*R(2)*R(3)+R(1)*R(2)*R(4)+R(1)*R(
3)*R(4)+R(2)*R(3)*R(4))*A
160 CP(0)=R(1)*R(2)*R(3)*R(4)*A
170 FOR I=1 TO 4 : RA(K,I)=R(I) : NEXT I:A(K)=A
180 GOSUB 1000
190 PRINT : PRINT : PRINT "Num. ";K
200 PRINT "Descomponer en factores el polinomio"
210 PRINT
220 PRINT EP$ : PRINT CP$
230 NEXT K
240 REM Soluciones
250 PRINT : PRINT
260 PRINT "***** S O L U C I O N E S *****"
270 FOR K=1 TO EJ
280 FOR I=1 TO 4
290 IF RA(K,I)>0 THEN R$(I)=STR$(-RA(K,I)) ELSE
R$(I)="+"+RIGHT$(STR$(RA(K,I)),1)
300 NEXT I
310 PRINT : PRINT "Num. ";K; " ";
320 IF A(K)=1 THEN PRINT " " : GOTO 350
330 IF A(K)=-1 THEN PRINT " -" : GOTO 350
340 PRINT STR$(A(K));
350 FOR I=1 TO 4
360 PRINT "(X";R$(I);")";
370 NEXT I
380 PRINT
390 NEXT K
400 END
1000 REM Subrutina escritura polinomio
1010 BL$=" " : CP$="" : EP$=""

```



```

1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=" "
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I)),
LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150
1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+"+RIGHT$(STR$(CP(0)),
LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>" " THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)
1180 FOR I=N TO 0 STEP -1
1190 CP$=CP$+CP$(I)
1200 NEXT I
1210 IF N=1 THEN RETURN
1220 FOR I=N TO 2 STEP -1
1230 IF CP(I)<>0 THEN EP$=EP$+LEFT$(BL$,LEN(CP$(
I))-2)+STR$(I)
1240 NEXT I
1250 RETURN

```

Guarde este programa con el nombre de FACTOPOL.BAS.

## 6.7 ECUACIONES DE CUARTO GRADO CON RAICES ENTERAS

Con este programa el ordenador elige al azar cuatro números enteros, comprendidos entre  $-5$  y  $5$ , y prepara un polinomio  $P(X)$  de cuarto grado que los tenga como raíces.

En el enunciado de cada ejercicio se exhibe la ecuación  $P(X) = 0$  y, al final, las cuatro soluciones de cada ejercicio.

En una ejecución del programa, resulta:

## ECUACIONES CON RAICES ENTERAS

Cuántos ejercicios? 10

Num. 1

Hallar las soluciones de la ecuación:

$$X^4 - 9X^3 + 24X^2 - 20X = 0$$

Num. 2

Hallar las soluciones de la ecuación:

$$X^4 + 2X^3 - 9X^2 - 2X + 8 = 0$$

Num. 3

Hallar las soluciones de la ecuación:

$$X^4 + 4X^3 - 19X^2 - 106X - 120 = 0$$

Num. 4

Hallar las soluciones de la ecuación:

$$X^4 - 2X^3 - 11X^2 + 12X = 0$$

Num. 5

Hallar las soluciones de la ecuación:

$$X^4 + 7X^3 - 9X^2 - 115X - 100 = 0$$

Num. 6

Hallar las soluciones de la ecuación:

$$X^4 + 5X^3 + 2X^2 - 8X = 0$$

Num. 7

Hallar las soluciones de la ecuación:

$$X^4 - 9X^3 + 22X^2 - 32 = 0$$

Num. 8

Hallar las soluciones de la ecuación:

$$X^4 + X^3 - 17X^2 + 15X = 0$$

Num. 9

Hallar las soluciones de la ecuación:

$$X^4 - 4X^3 - 16X^2 + 64X = 0$$

Num. 10

Hallar las soluciones de la ecuación:

$$X^4 + 11X^3 + 40X^2 + 48X = 0$$

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1      R1= 2      R2= 2      R3= 0      R4= 5

Num. 2      R1= 2      R2=-4      R3=-1      R4= 1

Num. 3      R1=-4      R2= 5      R3=-3      R4=-2

Num. 4	R1=-3	R2= 4	R3= 1	R4= 0
Num. 5	R1=-5	R2= 4	R3=-5	R4=-1
Num. 6	R1= 0	R2= 1	R3=-4	R4=-2
Num. 7	R1= 4	R2= 4	R3=-1	R4= 2
Num. 8	R1=-5	R2= 1	R3= 0	R4= 3
Num. 9	R1=-4	R2= 0	R3= 4	R4= 4
Num. 10	R1=-4	R2= 0	R3=-4	R4=-3

Este es el programa:

```

10 REM Ecuaciones de cuarto grado
20 CLS
30 PRINT "ECUACIONES CON RAICES ENTERAS" : PRINT
40 DEFINT A-Z
50 N=4
60 INPUT "Cuántos ejercicios";EJ
70 DIM R(4),RA(EJ,4),A(EJ),CP(4),CP$(4)
80 FOR K=1 TO EJ
90 FOR I=1 TO 4 : R(I)=INT(RND(1)*11)-5 : NEXT I
100 CP(4)=1
110 CP(3)=-(R(1)+R(2)+R(3)+R(4))
120 CP(2)=(R(1)*R(2)+R(1)*R(3)+R(1)*R(4)+R(2)*R(3)
)+R(2)*R(4)+R(3)*R(4))
130 CP(1)=-(R(1)*R(2)*R(3)+R(1)*R(2)*R(4)+R(1)*R(
3)*R(4)+R(2)*R(3)*R(4))
140 CP(0)=R(1)*R(2)*R(3)*R(4)
150 FOR I=1 TO 4 : RA(K,I)=R(I) : NEXT I
160 GOSUB 1000
170 PRINT : PRINT : PRINT "Num.";K
180 PRINT "Hallar las soluciones de la ecuación:"

```

```

190 PRINT
200 PRINT EP$ : PRINT CP$;" = 0"
210 NEXT K
220 REM Soluciones
230 PRINT : PRINT
240 PRINT "***** S O L U C I O N E S *****"
250 FOR K=1 TO EJ
260 PRINT : PRINT
270 PRINT "Num.";K;"    R1=";RA(K,1);"    R2=";RA(K
,2);"    R3=";RA(K,3);"    R4=";RA(K,4)
280 NEXT K
290 END
1000 REM Subrutina escritura polinomio
1010 BL$="      " : CP$="" : EP$=""
1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=" "
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I))
,LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150
1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+"+RIGHT$(STR$(CP(0))
,LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>" " THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)
1180 FOR I=N TO 0 STEP -1

```

Guarde este programa con el nombre de ECUACUAR.BAS.

## 6.8 OPERACIONES CON FRACCIONES ALGEBRAICAS

Con este programa se obtienen ejercicios de restar fracciones algebraicas, en las que los polinomios denominadores se pueden descomponer fácilmente en factores y, por tanto, calcular sin mucho trabajo el mínimo común denominador.

Las soluciones se dan en dos versiones: con el denominador descompuesto en factores y con las operaciones efectuadas. Veamos cómo funciona:

### FRACCIONES ALGEBRAICAS

Cuántos ejercicios ? 10

Num. 1 Calcula :

$$\frac{\overset{2}{x}}{x^2 - 4} - \frac{x-4}{x^2 - 4x + 4} - \frac{x+3}{2x-4} =$$

Num. 2 Calcula :

$$\frac{-3x^2}{x^2 - 1} - \frac{x-4}{x^2 + 2x + 1} - \frac{-2x+2}{x+1} =$$

Num. 3 Calcula :

$$\frac{\overset{2}{x}}{x^2 - 4} - \frac{x-1}{x^2 + 4x + 4} - \frac{-4x-3}{4x+8} =$$

Num. 4 Calcula :

$$\frac{\overset{2}{X}}{X^2 - 16} - \frac{X-1}{X^2 + 8X + 16} - \frac{-3X+1}{X-4} =$$

Num. 5 Calcula :

$$\frac{-\overset{2}{X}}{X^2 - 9} - \frac{X-2}{X^2 - 6X + 9} - \frac{-4X-1}{3X-9} =$$

Num. 6 Calcula :

$$\frac{-4\overset{2}{X}}{X^2 - 1} - \frac{X-1}{X^2 - 2X + 1} - \frac{3X-3}{2X-2} =$$

Num. 7 Calcula :

$$\frac{-3\overset{2}{X}}{X^2 - 9} - \frac{X+3}{X^2 + 6X + 9} - \frac{-2X+1}{X+3} =$$

Num. 8 Calcula :

$$\frac{\overset{2}{X}}{X^2 - 16} - \frac{X-3}{X^2 - 8X + 16} - \frac{2X+3}{-3X+12} =$$

Num. 9 Calcula :

$$\frac{-4x^2}{x^2-16} - \frac{x+2}{x^2+8x+16} - \frac{x+4}{3x+12}$$

Num. 10 Calcula :

$$\frac{x^2}{x^2-1} - \frac{x+1}{x^2-2x+1} - \frac{2x-2}{-4x+4}$$

\*\*\*\*\* SOLUCIONES \*\*\*\*\*

Num. 1

$$\frac{x^3 - 9x^2 + 8x + 28}{2(x+2)(x-2)^2} = \frac{x^3 - 9x^2 + 8x + 28}{2x^3 - 4x^2 - 8x + 16}$$

Num. 2

$$\frac{-x^3 - 6x^2 + 3x - 2}{(x-1)(x+1)^2} = \frac{-x^3 - 6x^2 + 3x - 2}{x^3 + x^2 - x - 1}$$

Num. 3

$$\frac{8x^3 + 7x^2 - 4x - 20}{4(x-2)(x+2)^2} = \frac{8x^3 + 7x^2 - 4x - 20}{4x^3 + 8x^2 - 16x - 32}$$



Num. 4

$$\frac{\begin{matrix} 3 & 2 \\ 2X & -4X & -53X+20 \end{matrix}}{\begin{matrix} 2 \\ -(x-4)(x+4) \end{matrix}} = \frac{\begin{matrix} 3 & 2 \\ 2X & -4X & -53X+20 \end{matrix}}{\begin{matrix} 3 & 2 \\ -X & -4X & +16X+64 \end{matrix}}$$

Num. 5

$$\frac{\begin{matrix} 3 & 2 \\ X & +7X & -39X+9 \end{matrix}}{\begin{matrix} 2 \\ 3(x+3)(x-3) \end{matrix}} = \frac{\begin{matrix} 3 & 2 \\ X & +7X & -39X+9 \end{matrix}}{\begin{matrix} 3 & 2 \\ 3X & -9X & -27X+81 \end{matrix}}$$

Num. 6

$$\frac{\begin{matrix} 3 & 2 \\ -11X & +9X & +3X-1 \end{matrix}}{\begin{matrix} 2 \\ 2(x+1)(x-1) \end{matrix}} = \frac{\begin{matrix} 3 & 2 \\ -11X & +9X & +3X-1 \end{matrix}}{\begin{matrix} 3 & 2 \\ 2X & -2X & -2X+2 \end{matrix}}$$

Num. 7

$$\frac{\begin{matrix} 3 & 2 \\ -X & -11X & -18X+18 \end{matrix}}{\begin{matrix} 2 \\ (x-3)(x+3) \end{matrix}} = \frac{\begin{matrix} 3 & 2 \\ -X & -11X & -18X+18 \end{matrix}}{\begin{matrix} 3 & 2 \\ X & +3X & -9X-27 \end{matrix}}$$

Num. 8

$$\frac{\begin{matrix} 3 & 2 \\ -5X & +12X & +35X+12 \end{matrix}}{\begin{matrix} 2 \\ -3(x+4)(x-4) \end{matrix}} = \frac{\begin{matrix} 3 & 2 \\ -5X & +12X & +35X+12 \end{matrix}}{\begin{matrix} 3 & 2 \\ -3X & +12X & +48X-192 \end{matrix}}$$

Num. 9

$$\frac{-13X^3 - 55X^2 + 22X + 88}{3(x-4)(x+4)^2} = \frac{-13X^3 - 55X^2 + 22X + 88}{3X^3 + 12X^2 - 48X - 192}$$

Num. 10

$$\frac{-6X^3 + 10X^2 + 10X + 2}{-4(x+1)(x-1)^2} = \frac{-6X^3 + 10X^2 + 10X + 2}{-4X^3 + 4X^2 + 4X - 4}$$

Este es el programa para obtener ejercicios de fracciones algebraicas:

```

10 REM Fracciones algebraicas
20 CLS
30 PRINT "FRACCIONES ALGEBRAICAS"
40 DEFINT A-Z
50 RA$=STRING$(15,196)
60 PRINT : PRINT : INPUT "Cuántos ejercicios ";EJ
70 DIM C(6),CP(3),CN(EJ,3),CD(EJ,2)
80 REM Escritura de los enunciados
90 FOR K=1 TO EJ
100 FOR I=1 TO 6
110 C(I)=INT(RND(1)*9)-4 : IF C(I)=0 THEN C(I)=1
120 NEXT I
130 IF C(5)=C(2)*C(3) THEN C(5)=C(5)+1
140 IF C(5)=0 THEN C(5)=1
150 PRINT : PRINT : PRINT
160 PRINT "Num.";K;" Calcula :" : PRINT
170 N=2 : CP(2)=C(3) : CP(1)=0 : CP(0)=0 :
GOSUB 1000
180 E1$=EP$ : P1$=CP$
190 N=1 : CP(1)=1 : CP(0)=-C(4) : GOSUB 1000
200 P2$=CP$
210 N=1 : CP(1)=C(5) : CP(0)=-C(6) : GOSUB 1000
220 P3$=CP$

```

```

230 PRINT TAB(2)E1$
240 PRINT TAB(2)P1$;TAB(11)P2$;TAB(22)P3$
250 PRINT LEFT$(RA$,7)+" - "+LEFT$(RA$,8)+" - "
+LEFT$(RA$,5)+" = "
260 N=2 : CP(2)=1 : CP(1)=0 : CP(0)=-C(1)*C(1):
GOSUB 1000
270 E1$=EP$ : P1$=CP$
280 N=2 : CP(2)=1 : CP(1)=-2*C(1) : CP(0)=C(1)*C(1)
) : GOSUB 1000
290 E2$=EP$ : P2$=CP$
300 N=1 : CP(1)=C(2) : CP(0)=-C(2)*C(1) :
GOSUB 1000
310 P3$=CP$
320 PRINT E1$;TAB(9)E2$
330 PRINT P1$;TAB(9)P2$;TAB(22)P3$
340 CN(K,3)=C(2)*C(3)-C(5)
350 CN(K,2)=-C(1)*C(2)*C(3)-C(2)+C(6)
360 CN(K,1)=-C(1)*C(2)+C(2)*C(4)+C(1)*C(1)*C(5)
370 CN(K,0)=C(1)*C(2)*C(4)-C(6)*C(1)*C(1)
380 CD(K,1)=C(1):CD(K,2)=C(2)
390 NEXT K
400 PRINT : PRINT : PRINT "***** SOLUCIONES *****"
410 FOR K=1 TO EJ
420 N=3 : FOR I=0 TO 3 : CP(I)=CN(K,I) : NEXT I
430 GOSUB 1000
440 PRINT : PRINT : PRINT "Num.";K : PRINT
450 PRINT EP$;TAB(19)EP$
460 PRINT CP$;TAB(19)CP$
470 PRINT RA$+" = ";
480 PRINT TAB(19)RA$
490 A$=MID$(STR$(CD(K,1)),2,1)
500 B$=STR$(CD(K,2)) : IF B$="-1" THEN B$="-"
510 IF B$=" 1" THEN B$=" "
520 S1$="+" : S2$="-"
530 IF CD(K,1)<0 THEN SWAP S1$,S2$
540 P$=B$+" (x "+S1$+A$+" ) (x "+S2$+A$+" )"
550 N=3 : CP(3)=CD(K,2) : CP(2)=-CD(K,1)*CD(K,2)
560 CP(1)=-CD(K,1)*CD(K,1)*CD(K,2) : CP(0)=CD(K,1)
^3*CD(K,2)

```

```

570 GOSUB 1000
580 PRINT TAB(2)SPACE$(LEN(P$))+ "2"; TAB(19)EP$
590 PRINT TAB(2)P$; TAB(19)CP$
600 NEXT K
610 END
1000 REM Subrutina escritura polinomio
1010 BL$="      " : CP$="" : EP$=""
1020 IF N=0 THEN CP$=STR$(CP(0)) : RETURN
1030 FOR I=0 TO N
1040 IF CP(I)=0 THEN CP$(I)=""
1050 NEXT I
1060 FOR I=1 TO N-1
1070 IF CP(I)=1 THEN CP$(I)="+ X " : GOTO 1110
1080 IF CP(I)=-1 THEN CP$(I)="- X " : GOTO 1110
1090 IF CP(I)<-1 THEN CP$(I)=STR$(CP(I))+ "X " :
GOTO 1110
1100 IF CP(I)>1 THEN CP$(I)="+"+RIGHT$(STR$(CP(I))
,LEN(STR$(CP(I)))-1)+"X "
1110 NEXT I
1120 IF CP(N)=1 THEN CP$(N)=" X " : GOTO 1150
1130 IF CP(N)=-1 THEN CP$(N)="- X " : GOTO 1150
1140 IF CP(N)<>0 THEN CP$(N)=STR$(CP(N))+ "X "
1150 IF CP(0)>0 THEN CP$(0)="+"+RIGHT$(STR$(CP(0))
,LEN(STR$(CP(0)))-1)
1160 IF CP(0)<0 THEN CP$(0)=STR$(CP(0))
1170 IF CP$(1)<>" " THEN CP$(1)=LEFT$(CP$(1),LEN(
CP$(1))-1)
1180 FOR I=N TO 0 STEP -1
1190 CP$=CP$+CP$(I)
1200 NEXT I
1210 IF N=1 THEN RETURN
1220 FOR I=N TO 2 STEP -1
1230 IF CP(I)<>0 THEN EP$=EP$+LEFT$(BL$,LEN(CP$(
I))-2)+STR$(I)
1240 NEXT I
1250 RETURN

```

Guarde el programa en el disco con el nombre de FRACALGE.BAS.

# 7

## DETERMINANTES Y SISTEMAS

### 7.1 DETERMINANTES

El ordenador pregunta primero por el orden de las matrices que se van a manejar y luego por el número de ejercicios que se desean.

Veamos qué ocurre cuando se ejecuta el programa:

Determinantes de orden ? 2

Cuántos ejercicios ? 6

Num. 1

$$\begin{vmatrix} 9 & 12 \\ 12 & 16 \end{vmatrix}$$

Num. 4

$$\begin{vmatrix} -7 & -2 \\ -8 & -2 \end{vmatrix}$$

Num. 2

$$\begin{vmatrix} 11 & 16 \\ 14 & 20 \end{vmatrix}$$

Num. 5

$$\begin{vmatrix} -4 & -5 \\ 1 & 1 \end{vmatrix}$$

Num. 3

$$\begin{vmatrix} -5 & -6 \\ -4 & -4 \end{vmatrix}$$

Num. 6

$$\begin{vmatrix} 1 & 2 \\ -2 & 0 \end{vmatrix}$$

Soluciones de los determinantes

Num. 1     $\det = 0$

Num. 2     $\det = -4$

Num. 3     $\det = -4$

Num. 4     $\det = -2$

Num. 5     $\det = 1$

Num. 6     $\det = 4$

## Determinantes de orden ? 4

### Cuántos ejercicios ? 6

Num. 1

$$\begin{vmatrix} 2 & 19 & -2 & 1 \\ 7 & -15 & 2 & -1 \\ 8 & -4 & 2 & 1 \\ 5 & 1 & 1 & 1 \end{vmatrix}$$

Num. 4

$$\begin{vmatrix} -7 & 3 & 8 & 5 \\ 21 & 25 & 14 & 11 \\ 10 & -12 & -14 & -9 \\ 11 & -15 & -17 & -11 \end{vmatrix}$$

Num. 2

$$\begin{vmatrix} -9 & -9 & 26 & 31 \\ 1 & -3 & 4 & 5 \\ 0 & 4 & -4 & -5 \\ -3 & 1 & 1 & 1 \end{vmatrix}$$

Num. 5

$$\begin{vmatrix} -7 & 15 & 4 & 7 \\ -7 & 25 & 8 & 13 \\ -4 & 0 & 0 & -1 \\ -3 & 3 & 1 & 1 \end{vmatrix}$$

Num. 3

$$\begin{vmatrix} -7 & 3 & -2 & -1 \\ 9 & 23 & 2 & 7 \\ -6 & -14 & -2 & -5 \\ -3 & -9 & -1 & -3 \end{vmatrix}$$

Num. 6

$$\begin{vmatrix} 26 & 3 & 14 & 11 \\ -11 & -23 & 6 & 3 \\ 6 & -2 & 4 & 3 \\ 5 & 3 & 1 & 1 \end{vmatrix}$$

### Soluciones de los determinantes

Num. 1	det = -3	Num. 4	det = -2
Num. 2	det = 4	Num. 5	det = 2
Num. 3	det = -4	Num. 6	det = -5

## Determinantes de orden ? 10

### Cuántos ejercicios ? 3

Num. 1

$$\begin{vmatrix} -31 & -15 & 42 & -1 & 0 & 11 & -4 & 13 & 4 & 1 \\ -23 & 27 & 6 & -21 & 18 & 19 & 4 & 11 & 4 & 1 \\ 14 & 10 & 14 & -9 & -2 & -25 & -16 & 1 & -2 & 1 \\ -1 & 13 & 11 & -11 & 0 & -27 & -6 & 7 & 0 & 3 \\ 68 & 38 & -32 & 10 & 14 & -17 & -8 & -7 & -4 & 1 \\ 11 & 23 & 23 & -11 & 45 & -1 & -28 & 37 & 8 & 5 \\ -22 & -10 & 6 & 2 & -10 & 10 & 12 & -5 & 0 & -1 \\ -3 & -7 & -15 & -9 & -9 & 3 & 5 & -11 & -2 & -3 \\ 16 & 6 & -4 & 10 & -4 & -10 & 0 & -4 & -2 & 1 \\ 21 & 7 & -7 & 13 & -7 & -13 & 1 & -7 & -3 & 1 \end{vmatrix}$$

Num. 2

-37	-7	-2	9	18	-3	14	11	14	17
-17	15	8	-1	-6	21	8	1	2	3
12	28	18	-3	20	15	-4	-9	-8	-11
7	-29	-9	15	20	21	-2	-7	-6	-9
-14	14	8	0	6	9	6	3	4	5
1	-27	-15	23	21	33	-8	-3	-6	-9
8	-6	-4	2	-4	4	-4	-3	-4	-5
7	-3	-3	1	-3	-1	-5	-3	-4	-5
-12	6	2	0	-4	8	10	10	10	13
-15	9	3	-1	-7	9	13	13	13	17

Num. 3

30	15	-12	-3	-22	21	0	3	18	23
35	27	-14	33	10	11	-4	7	2	3
24	16	-16	23	-4	13	-2	5	10	13
-1	21	-1	23	4	15	-2	1	0	-1
36	10	-30	-10	-20	17	4	1	18	23
17	27	-25	3	-7	15	-8	5	12	15
30	32	-34	16	-2	18	-4	5	12	15
23	15	-33	-15	-27	21	-1	1	20	25
4	2	-6	-4	-6	4	0	0	4	5
9	3	-9	-7	-9	5	-1	1	7	9

### Soluciones de los determinantes

Num. 1     det = -2

Num. 2     det = 2

Num. 3     det = -5

El orden de las matrices está limitado únicamente por las dimensiones de la pantalla o del papel de la impresora.

En realidad existe una limitación teórica, pero que en la práctica no se presenta, y es la debida a la línea 330 del programa que es la que

escribe los elementos de la matriz. Pues bien, para que haya una perfecta alineación de los elementos, se escriben siempre los tres últimos caracteres (incluido el signo), completando con espacios, si tuviera menos. De modo que para los números de dos cifras no habrá problema, ni para los positivos de tres cifras; pero un número negativo de tres cifras, como  $-123$ , aparecerá externamente escrito como  $123$ , y por tanto, el determinante de la matriz presentada no coincidirá con el valor del determinante dado en la solución.

Afortunadamente el lector no debe preocuparse por esta limitación. En la práctica podrá obtener, si lo desea, determinantes de matrices cuadradas de orden 20, ... ¡con solución garantizada!

No se piense que el ordenador calcula realmente el valor de los determinantes. Nada de eso: los prepara para que tengan un valor prefijado. Veamos cómo lo hace:

El algoritmo para preparar matrices con determinantes de valor prefijado está basado en las siguientes ideas:

1. El determinante de un producto de matrices es el producto de los determinantes:

$$\det(A.B) = \det(A).\det(B)$$

2. El determinante de una matriz triangular es el producto de los elementos de la diagonal principal.

Veamos, con un ejemplo, cómo se “fabrica” una matriz  $C$  de orden 3, cuyo determinante queremos que valga 5:

$$\begin{matrix} \begin{pmatrix} 1 & -1 & 3 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix} & \begin{pmatrix} 5 & 0 & 0 \\ -1 & 1 & 0 \\ 3 & -3 & 1 \end{pmatrix} & = & \begin{pmatrix} 15 & -10 & 3 \\ 2 & -2 & 1 \\ 3 & -3 & 1 \end{pmatrix} \\ A & B & & C \end{matrix}$$

La matriz  $A$  es triangular, con “unos” en la diagonal principal, por tanto, su determinante vale 1. La  $B$  también es triangular y tiene “unos”



en la diagonal principal, excepto en la posición (1,1), que tiene precisamente un 5; su determinante valdrá 5. En consecuencia, el determinante de  $C$  valdrá  $1 \times 5 = 5$ .

Los elementos que están por encima de la diagonal principal en  $A$  y por debajo en  $B$ , que son irrelevantes a efectos del valor del determinante de  $C$ , son importantes porque serán los que hagan variar el "aspecto final" de la matriz  $C$ . Estos números han de cumplir ciertas condiciones: no ser muy grandes, para que los elementos de  $C$  tampoco lo sean; no ser nulos, porque si no aparecen muchos ceros en  $C$ ; y, por supuesto, ser aleatorios.

Finalmente, observemos que este algoritmo conserva la última columna de  $A$  y la última fila de  $B$ . Para evitar esto, la matriz  $C$ , que finalmente se utiliza, es la que resulta de cambiar la última columna por la suma de ella con la penúltima y, después, la última fila por la suma de ella con la anterior. Con todos estos cambios el determinante se conserva:

$$C = \begin{pmatrix} 15 & -10 & 3 \\ 2 & -2 & 1 \\ 3 & -3 & 1 \end{pmatrix} \rightarrow \begin{pmatrix} 15 & -10 & -7 \\ 2 & -2 & -1 \\ 3 & -3 & -2 \end{pmatrix} \rightarrow \begin{pmatrix} 15 & -10 & -7 \\ 2 & -2 & -1 \\ 5 & -5 & -3 \end{pmatrix} = C$$

Este es el programa que prepara ejercicios de determinantes:

```

10 REM Determinantes
20 CLS
30 DEFINT A-Z
40 INPUT "Determinantes de orden ";N : PRINT
50 INPUT "Cuántos ejercicios ";EJ
60 DIM X(EJ),A(N,N),B(N,N),C(N,N)
70 REM Generación de la matriz
80 FOR I=1 TO N : A(I,I)=1 : B(I,I)=1:NEXT I
90 FOR K=1 TO EJ
100 FOR I=1 TO N-1
110 FOR J=I+1 TO N
120 A(I,J)=INT(RND(1)*5)*2-3
130 B(J,I)=INT(RND(1)*5)*2-3
140 NEXT J
150 NEXT I

```

```

160 X(K)=INT(RND(1)*11-5)
170 B(1,1)=X(K)
180 FOR I=1 TO N
190 FOR J=1 TO N
200 C(I,J)=0
210 FOR L =1 TO N
220 C(I,J)=C(I,J)+A(I,L)*B(L,J)
230 NEXT L
240 NEXT J
250 NEXT I
260 FOR I=1 TO N : C(I,N)=C(I,N)+C(I,N-1) : NEXT I
270 FOR J=1 TO N : C(N,J)=C(N,J)+C(N-1,J) : NEXT J
280 REM Escritura de la matriz
290 PRINT : PRINT : PRINT "Num. "; K : PRINT
300 FOR I=1 TO N
310 PRINT CHR$(179);
320 FOR J=1 TO N
330 PRINT RIGHT$(" "+STR$(C(I,J)),3); " ";
340 NEXT J
350 PRINT CHR$(179)
360 NEXT I
370 NEXT K
380 REM Escritura de la solución
390 PRINT : PRINT
400 PRINT "Soluciones de los determinantes" : PRINT
410 FOR K=1 TO EJ
420 PRINT "Num. "; K; "   det = "; X(K)
430 NEXT K
440 END

```

En las líneas 310 y 350 figura CHR\$(179), que es la rayita vertical. Si su ordenador tiene esa rayita como carácter gráfico, puede sustituir CHR\$(179) por "|".

Conserve este programa en el disco con el nombre de DETERMIN.BAS.

## 7.2 SISTEMAS DE ECUACIONES LINEALES

Este programa presenta tantos sistemas de ecuaciones como se deseen. Todos los sistemas son determinados y tienen el mismo número

de ecuaciones que de incógnitas. Este número no puede ser superior a 6, pero es muy sencillo superar este límite: basta añadir más incógnitas en la línea 80 del programa.

Veamos qué pasa cuando se ejecuta el programa:

Sistemas de cuántas ecuaciones ? 3

Cuántos ejercicios ? 10

Num. 1

$$\begin{aligned} 10x &+ z = 14 \\ 10x - 8y - 5z &= 14 \\ 13x - 11y - 7z &= 18 \end{aligned}$$

Num. 6

$$\begin{aligned} -2x - 12y - 9z &= -40 \\ -4x - 2y - z &= -2 \\ -7x - 5y - 3z &= -9 \end{aligned}$$

Num. 2

$$\begin{aligned} x - z &= 4 \\ 2x + z &= 5 \\ 5x - y + z &= 18 \end{aligned}$$

Num. 7

$$\begin{aligned} 8x + 4y + 3z &= 39 \\ 4x + 4y + 7z &= 11 \\ 5x + 5y + 9z &= 13 \end{aligned}$$

Num. 3

$$\begin{aligned} 8x - 8y - 11z &= 37 \\ -4x + 16y + 21z &= -51 \\ -5x + 19y + 25z &= -61 \end{aligned}$$

Num. 8

$$\begin{aligned} -3x - 4y - z &= 32 \\ -18x - 4y + z &= 84 \\ -21x - 5y + z &= 100 \end{aligned}$$

Num. 4

$$\begin{aligned} 3x - 8y - 11z &= 57 \\ 8x + 16y + 21z &= -111 \\ 9x + 19y + 25z &= -132 \end{aligned}$$

Num. 9

$$\begin{aligned} 5x - 4y - 5z &= -18 \\ -6x - 8y - 11z &= 30 \\ -5x - 5y - 7z &= 24 \end{aligned}$$

Num. 5

$$\begin{aligned} -9x + 14y + 19z &= 28 \\ -4x + 4y + 5z &= 11 \\ -7x + 7y + 9z &= 20 \end{aligned}$$

Num. 10

$$\begin{aligned} 10x - 4y - 7z &= -15 \\ 6x - z &= -21 \\ 3x + y + z &= -16 \end{aligned}$$

\*\*\*\*\* SOLUCIONES \*\*\*\*\*

- |                        |                          |
|------------------------|--------------------------|
| Num. 1 ( 1 , -3 , 4 )  | Num. 6 ( -1 , 2 , 2 )    |
| Num. 2 ( 3 , -4 , -1 ) | Num. 7 ( 4 , 4 , -3 )    |
| Num. 3 ( 2 , -4 , 1 )  | Num. 8 ( -4 , -4 , -4 )  |
| Num. 4 ( 0 , -3 , -3 ) | Num. 9 ( -4 , 2 , -2 )   |
| Num. 5 ( -3 , -4 , 3 ) | Num. 10 ( -4 , -1 , -3 ) |

Probemos otra vez:

Sistemas de cuántas ecuaciones ? 4

Cuántos ejercicios ? 10

Num. 1

$$\begin{array}{rclcl}
 & +13y & - 2z & + & u & = & 9 \\
 7x & - 9y & + 2z & - & u & = & 23 \\
 6x & - 2y & + 2z & + & u & = & 18 \\
 3x & + & y & + & z & + & u & = & 9
 \end{array}$$

Num. 2

$$\begin{array}{rclcl}
 -15x & + & y & - 4z & - 3u & = & 18 \\
 -19x & - 7y & -16z & -11u & = & 50 \\
 -8x & & & - 8z & - 5u & = & 32 \\
 -11x & - & y & -11z & - 7u & = & 42
 \end{array}$$

Num. 3

$$\begin{array}{rclcl}
 26x & +17y & -12z & -15u & = & -122 \\
 -13x & -13y & +14z & +19u & = & 57 \\
 -2x & - 4y & + 4z & + 5u & = & 14 \\
 -5x & - 7y & + 7z & + 9u & = & 28
 \end{array}$$

Num. 4

$$\begin{array}{rclcl}
 -3x & -13y & + 6z & + 3u & = & -7 \\
 5x & +11y & - 8z & - 5u & = & 1 \\
 4x & +10y & - 8z & - 5u & = & -2 \\
 5x & +13y & -11z & - 7u & = & -5
 \end{array}$$

Num. 5

$$\begin{aligned}-18x - 3y - 10z - 7u &= 106 \\ -9x + 5y + 14z + 11u &= -18 \\ 6x + 8y + 10z + 7u &= -68 \\ 3x + 5y + 7z + 5u &= -42\end{aligned}$$

Num. 6

$$\begin{aligned}-14x + 17y - 4z - 5u &= -1 \\ 5x - 5y + 2z - u &= 6 \\ -6y + 2z + 3u &= -2 \\ -x - 9y + 3z + 5u &= -4\end{aligned}$$

Num. 7

$$\begin{aligned}2x + 11y - 2z - 5u &= 39 \\ -9x - 11y + 3u &= -67 \\ 6x - 8y + 4z + 7u &= -1 \\ 7x - 11y + 5z + 9u &= -6\end{aligned}$$

Num. 8

$$\begin{aligned}12x + y + u &= -26 \\ -7x + 3y + 8z + 13u &= -24 \\ + 2y + 4z + 7u &= -20 \\ x + 3y + 5z + 9u &= -28\end{aligned}$$

Num. 9

$$\begin{aligned}9x + 3y - 2z - 5u &= -42 \\ -5x + y + 2z + 3u &= 20 \\ -4x - 2y + 4z + 7u &= 31 \\ -5x - 3y + 5z + 9u &= 40\end{aligned}$$

Num. 10

$$\begin{aligned}5x + 7y + 2z + 3u &= -8 \\ -x + y + 4z + 5u &= 2 \\ 8x + 6y + 10z + 13u &= 23 \\ 11x + 9y + 13z + 17u &= 27\end{aligned}$$

\*\*\*\*\* SOLUCIONES \*\*\*\*\*

Num. 1 ( 4 , 1 , 0 , -4 )  
Num. 2 ( 0 , 2 , -4 , 0 )  
Num. 3 ( -3 , -4 , 3 , -4 )  
Num. 4 ( 2 , 1 , 0 , 4 )  
Num. 5 ( -4 , -2 , 0 , -4 )  
Num. 6 ( -1 , -3 , -4 , -4 )  
Num. 7 ( 4 , 2 , 3 , -3 )  
Num. 8 ( -2 , -2 , -4 , 0 )  
Num. 9 ( -3 , -2 , 2 , 1 )  
Num. 10 ( 3 , -4 , 1 , 1 )

El programa no resuelve sistemas, sino que los prepara eligiendo al azar la matriz  $C$  de los coeficientes y la solución, y calculando posteriormente los términos independientes. Con el algoritmo del programa de determinantes construye la matriz  $C$  de determinante no nulo; de esta manera asegura que el sistema propuesto sea determinado.

```
10 REM Sistemas de ecuaciones lineales
20 DEFINT A-Z
30 CLS
40 INPUT "Sistemas de cuántas ecuaciones ";N
50 PRINT : INPUT "Cuántos ejercicios ";EJ
60 PRINT : PRINT
70 DIM A(N,N),B(N,N),C(N,N),C$(N,N),X(EJ,N),L$(N)
,D(N)
80 DATA x,y,z,u,v,w
90 REM Generación de la matriz de los coeficientes
100 FOR I=1 TO N
110 A(I,I)=1 : B(I,I)=1 : READ L$(I)
120 NEXT I
130 FOR K=1 TO EJ
140 FOR I=1 TO N-1
```

```

150 FOR J=I+1 TO N
160 A(I,J)=INT(RND(1)*5)*2-3
170 B(J,I)=INT(RND(1)*4)*2-3
180 NEXT J
190 NEXT I
200 B(1,1)=INT(RND(1)*3)+3
210 FOR I=1 TO N
220 FOR J=1 TO N
230 C(I,J)=0
240 FOR T=1 TO N
250 C(I,J)=C(I,J)+A(I,T)*B(T,J)
260 NEXT T
270 NEXT J
280 NEXT I
290 FOR I=1 TO N : C(I,N)=C(I,N)+C(I,N-1) : NEXT I
300 FOR J=1 TO N : C(N,J)=C(N,J)+C(N-1,J) : NEXT J
310 FOR J=1 TO N
320 X(K,J)=INT(RND(1)*9)-4
330 NEXT J
340 REM Cálculo de los terminos independientes
350 FOR I=1 TO N
360 D(I)=0
370 FOR J=1 TO N
380 D(I)=D(I)+C(I,J)*X(K,J)
390 NEXT J
400 NEXT I
410 REM Preparación para la escritura
420 FOR I=1 TO N
430 IF C(I,1)=1 THEN C$(I,1)=" "+L$(1)
440 IF C(I,1)=-1 THEN C$(I,1)=" -"+L$(1)
450 IF C(I,1)>9 OR C(I,1)<-9 THEN C$(I,1)=STR$(C(I,1))+L$(1) : GOTO 480
460 IF C(I,1)>1 OR C(I,1)<-1 THEN C$(I,1)=" " + STR$(C(I,1))+L$(1)
470 IF C(I,1)=0 THEN C$(I,1)=" "
480 NEXT I
490 FOR I=1 TO N
500 FOR J=2 TO N
510 IF C(I,J)=1 THEN C$(I,J)="+ "+L$(J)

```

```

520 IF C(I,J)=-1 THEN C$(I,J)="- "+L$(J)
530 IF C(I,J)>9 THEN C$(I,J)="+"+RIGHT$(STR$(C(I
,J)),2)+L$(J) : GOTO 580
540 IF C(I,J)<-9 THEN C$(I,J)="-"+RIGHT$(STR$(C(
I,J)),2)+L$(J) : GOTO 580
550 IF C(I,J)>1 THEN C$(I,J)="+"+STR$(C(I,J))+L$
(J)
560 IF C(I,J)<-1 THEN C$(I,J)="- "+RIGHT$(STR$(C
(I,J)),1)+L$(J)
570 IF C(I,J)=0 THEN C$(I,J)=" "
580 NEXT J
590 NEXT I
600 REM Escritura del sistema
610 PRINT : PRINT "Num. ";K : PRINT
620 FOR I=1 TO N
630 FOR J=1 TO N
640 PRINT C$(I,J)+" ";
650 NEXT J
660 PRINT "=";D(I)
670 NEXT I
680 PRINT
690 NEXT K
700 REM Escritura de la solución
710 PRINT : PRINT
720 PRINT "***** SOLUCIONES *****"
730 FOR K=1 TO EJ
740 PRINT : PRINT "Num. ";K; " (";
750 FOR J=1 TO N-1
760 PRINT X(K,J); ", ";
770 NEXT J
780 PRINT X(K,N); ") "
790 NEXT K
800 END

```

Guarde este programa con el nombre de SISTEMAS.BAS.



# 8

## ESTADISTICA

### 8.1 DISTRIBUCION NORMAL

Presentamos aquí un programa muy flexible para simular *distribuciones normales*, que permite escoger el número de ejercicios, el número de datos de cada ejercicio y los intervalos de los que se obtienen aleatoriamente *la media y la desviación típica* de cada distribución. Para que los resultados salgan bien colocados en forma de tabla, es necesario elegir también el número de cifras enteras y decimales de los datos.

Imagínese que necesita cuatro ejercicios que simulen las calificaciones otorgadas por un profesor, que aprecie hasta las décimas de punto.

Suponiendo que usted conoce que la media de ese profesor suele variar entre 4 y 5, y que la desviación típica varía entre 1 y 2, puede ejecutar el programa así:

D I S T R I B U C I O N      N O R M A L

Número de ejercicios? 4

Número de datos en cada ejercicio ( 50 ó 100 )? 50

Media aleatoriamente comprendida entre M1 y M2

M1 , M2 ? 4,5

Desv. típ. aleatoriamente comprendida entre D1 y D2

D1 , D2 ? 1,2

Número de cifras enteras ( 1 , 2 , 3 , 4 ó 5 )? 2(\*)

Número de cifras decimales ( 0 , 1 ó 2 )? 1

---

(\*) Hay que prever la posibilidad de que el ordenador genere notas de dos cifras enteras, como 10.0, por ejemplo.

### Num. 1

4.6	3.9	4.7	3.4	4.6
4.3	0.7	1.6	1.6	5.3
5.5	1.8	4.1	6.6	4.0
4.3	3.6	6.7	2.9	3.7
3.6	7.2	7.3	1.9	4.0
6.4	5.4	5.6	4.8	7.8
7.3	4.5	6.0	4.5	2.8
4.0	2.6	6.7	5.5	7.3
6.4	5.9	5.3	2.5	7.2
5.5	2.0	4.8	6.7	7.3

### Num. 2

5.9	5.6	3.2	6.1	4.2
4.8	5.5	3.8	7.1	5.3
4.7	6.2	3.8	6.2	3.2
3.3	4.0	5.1	6.3	6.1
4.5	5.4	5.2	4.9	6.4
3.7	4.5	2.9	4.1	4.6
4.0	3.3	6.5	5.7	3.8
6.2	4.8	2.7	3.9	3.9
4.5	6.5	5.4	5.1	4.4
3.0	3.8	5.5	4.7	4.1

### Num. 3

4.9	2.8	5.6	5.2	5.9
5.8	6.0	3.1	2.8	4.7
3.8	4.4	4.5	2.3	2.0
4.5	5.2	6.5	5.0	3.6
4.7	4.9	5.5	7.5	4.5
3.8	1.8	4.3	4.8	6.0
4.5	3.0	5.4	5.6	4.6
5.0	4.9	4.1	6.5	4.4
4.8	2.9	4.3	4.2	5.0
6.6	5.5	4.3	7.2	4.7

### Num. 4

3.0	6.1	6.1	4.4	7.8
6.9	6.5	4.5	4.0	4.5
7.4	5.1	6.0	6.2	5.2

6.1	3.2	8.4	5.2	4.0
6.1	2.0	2.1	5.9	5.9
3.3	4.8	4.7	6.1	6.8
4.4	5.8	6.8	1.7	8.0
7.3	6.2	0.7	6.7	5.6
5.9	2.0	2.2	7.9	4.4
1.9	3.6	0.0	5.1	4.0

Num. 1	MEDIA= 4.7	DESV.TIP.= 1.7
Num. 2	MEDIA= 4.6	DESV.TIP.= 1.1
Num. 3	MEDIA= 4.9	DESV.TIP.= 1.5
Num. 4	MEDIA= 4.8	DESV.TIP.= 1.8

Si usted quiere que todos los ejercicios tengan la misma media y desviación típica, por ejemplo  $M=5$  y  $D=1$ , no tiene más que reducir los respectivos intervalos a un punto:

## D I S T R I B U C I O N      N O R M A L

Número de ejercicios? 4

Número de datos en cada ejercicio ( 50 ó 100 )? 50

Media aleatoriamente comprendida entre M1 y M2

M1 , M2 ? 5,5

Desv. típ. aleatoriamente comprendida entre D1 y D2

D1 , D2 ? 1,1

Número de cifras enteras (1 , 2 , 3 , 4 ó 5 )? 2

Número de cifras decimales ( 0 , 1 ó 2 )? 1

Num. 1

4.9	4.5	5.0	4.3	4.9
4.8	2.6	3.2	3.2	5.3
5.5	3.3	4.6	6.1	4.6
4.8	4.4	6.2	4.0	4.4
4.4	6.5	6.5	3.3	4.6
6.0	5.4	5.5	5.1	6.8
6.5	4.9	5.8	4.9	3.9
4.6	3.8	6.2	5.5	6.5
6.0	5.7	5.3	3.7	6.5
5.5	3.4	5.1	6.2	6.5

### Num. 2

6.1	5.9	3.7	6.4	4.6
5.2	5.8	4.3	7.3	5.7
5.1	6.4	4.3	6.4	3.7
3.8	4.4	5.4	6.5	6.4
4.9	5.7	5.6	5.3	6.6
4.2	4.9	3.4	4.5	5.0
4.5	3.8	6.7	6.0	4.3
6.5	5.2	3.2	4.4	4.3
4.9	6.8	5.7	5.5	4.8
3.5	4.3	5.9	5.0	4.5

### Num. 3

5.0	3.6	5.4	5.2	5.7
5.6	5.7	3.8	3.6	4.9
4.3	4.7	4.7	3.3	3.1
4.8	5.2	6.0	5.1	4.1
4.9	5.0	5.4	6.7	4.8
4.2	2.9	4.6	4.9	5.7
4.7	3.7	5.3	5.5	4.8
5.0	5.0	4.4	6.1	4.7
5.0	3.7	4.6	4.5	5.1
6.1	5.4	4.6	6.5	4.9

### Num. 4

4.0	5.7	5.7	4.8	6.6
6.1	5.9	4.8	4.5	4.9
6.5	5.2	5.7	5.8	5.2
5.7	4.1	7.0	5.2	4.5
5.7	3.4	3.5	5.6	5.6
4.2	5.0	5.0	5.7	6.1
4.8	5.5	6.1	3.3	6.8
6.4	5.8	2.7	6.1	5.4
5.6	3.4	3.6	6.7	4.8
3.4	4.3	2.3	5.2	4.6

Num. 1	MEDIA= 5.0	DESV.TIP.= 1.0
Num. 2	MEDIA= 5.0	DESV.TIP.= 1.0
Num. 3	MEDIA= 5.0	DESV.TIP.= 1.0
Num. 4	MEDIA= 5.0	DESV.TIP.= 1.0

He aquí el programa que genera distribuciones normales:

```
10 REM Distribución normal
20 CLS
30 PRINT "D I S T R I B U C I O N      N O R M A L"
40 DEFSNG A-Z
50 PRINT : PRINT
60 INPUT "Número de ejercicios";EJ%
70 DIM M(EJ%),D(EJ%)
80 U$(0)="" : U$(1)="#" : U$(2)="##" : U$(3)="###"
: U$(4)="####" : U$(5)="#####"
90 PRINT : INPUT "Número de datos en cada ejercic
io ( 50 ó 100 )";DA%
100 PRINT : PRINT "Media aleatoriamente comprendid
a entre M1 y M2" : PRINT
110 INPUT "M1 , M2 ";M1,M2
120 PRINT : PRINT "Desviación típica aleatoriamen
te comprendida entre D1 y D2" : PRINT
130 INPUT "D1 , D2 ";D1,D2
140 PRINT : INPUT "Número de cifras enteras ( 1 , 2
,3 , 4 ó 5 )";CE%
150 PRINT : INPUT "Número de cifras decimales ( 0
, 1 ó 2 )";CD%
160 U$=U$(CE%)+". "+U$(CD%)
170 RE%=10^CD% : DEF FN RED(Z)=INT(RE%*Z+.5)/RE%
180 FOR K%=1 TO EJ%
190 PRINT : PRINT : PRINT "Num.";K% : PRINT
200 M(K%)=M1+(M2-M1)*RND(1) : M(K%)=FN RED(M(K%))
210 D(K%)=D1+(D2-D1)*RND(1) : D(K%)=FN RED(D(K%))
220 FOR I%=1 TO DA%/5 : FOR J%=1 TO 5
230 GOSUB 330
240 X=M(K%)+D(K%)*Z
250 PRINT USING U$;X; : PRINT " ";
260 NEXT J% : PRINT : NEXT I%
270 NEXT K%
280 PRINT : PRINT
290 FOR K%=1 TO EJ%
300 PRINT "Num.";K%;" MEDIA=";:PRINT USING U$;M(
K%);:PRINT " DESV.TIP.=";:PRINT USING U$;D(K%)
```

```

310 NEXT K%
320 END
330 REM Subrutina N(0,1)
340 S=0
350 FOR H%=1 TO 12
360 S=S+RND(1)
370 NEXT H%
380 Z=S-6
390 RETURN

```

Guárdelo con el nombre de DISNORMA.BAS.

El corazón del programa es la subrutina 330, que proporciona valores  $Z$  de la distribución estándar  $N(0,1)$ , que la línea 240 se encarga de transformar en números  $X$ , distribuidos normalmente  $N(M(K\%), D(K\%))$ .

Debe quedar claro que el programa anterior no calcula ni la media ni la desviación típica de cada ejercicio, sino que, cada vez, prepara una distribución “que tenga” la media y la desviación prefijadas. Por tanto, la media y desviación típica reales, las que tiene que calcular el alumno, no tienen por qué coincidir exactamente con las ofrecidas en la solución. Sin embargo, podemos esperar que el error sea razonablemente pequeño. Para que el lector pueda comprobar esto, y decidir por sí mismo el grado de confianza que le merece el programa DISNORMA.BAS, hemos preparado el programa TESTSIMU.BAS que genera 100 datos con media y desviación elegidas por el usuario y calcula efectivamente la media y desviación típica de la distribución generada, de modo que se puedan comparar ambas parejas de números. Veamos cómo funciona:

```

Media , Desv.tip. ? 5,1
Media = 5.09 Desv.tip. = 1.02

Media , Desv.tip. ? 5,1
Media = 4.98 Desv.tip. = 0.97

Media , Desv.tip. ? 5,1
Media = 4.91 Desv.tip. = 0.91

Media , Desv.tip. ? 5,1
Media = 5.21 Desv.tip. = 1.12

```

```
Media , Desv.tip. ? 5,1
Media = 4.98 Desv.tip. = 0.94
```

```
Media , Desv.tip. ? 5,1
Media = 4.95 Desv.tip. = 1.03
```

Observe que para la media, el error cometido es del orden del 5 % (en algún caso puede llegar hasta el 10 %). El error cometido con el caso de la desviación típica es mayor, entorno al 10 % (en algún caso puede llegar hasta el 20 %). Estas observaciones habrán de tenerse en cuenta a la hora de calificar las soluciones dadas por los alumnos.

```
10 REM Test simulación distribución normal
20 CLS
30 INPUT "Media , Desv.tip.";M,D
40 SX=0 : XX=0
50 FOR I=1 TO 100
60 GOSUB 150
70 X=M+D*Z
80 SX=SX+X : XX=XX+X*X
90 NEXT I
100 M=SX/100 : D=SQR(XX/100-M*M)
110 PRINT "Media = ";INT(M*100+.5)/100;" Desv.
tip. = ";INT(D*100+.5)/100
120 PRINT
130 GOTO 30
140 END
150 REM Subrutina N(0,1)
160 S=0
170 FOR H=1 TO 12
180 S=S+RND(1)
190 NEXT H
200 Z=S-6
210 RETURN
```

Guarde este programa con el nombre de TESTSIMU.BAS.

## 8.2 DISTRIBUCION BIDIMENSIONAL

Es posible simular una *distribución bidimensional* ( $X, Y$ ), de modo que ambas variables estadísticas se distribuyan normalmente, con *media* y *desviación típica* prefijadas, y con *coeficiente de correlación* también prefijado.

El programa DISTRIBI.BAS genera tantos ejercicios, como necesitamos, de distribuciones bidimensionales, y ofrece al final, como solución, la media y desviación de ambas variables, así como el coeficiente de correlación.

Supongamos que se necesitan ocho ejercicios de la distribución PESO/TALLA (peso en kg, talla, en cm). Podemos proceder así:

### DISTRIBUCION BIDIMENSIONAL NORMAL

Número de ejercicios? 8

Número de datos en cada ejercicio ( 20 ó 50 )? 20

Media de las  $x$  comprendida entre MX1 y MX2

MX1 , MX2 ? 60,70

Desv. típ. de las  $x$  comprendida entre DX1 y DX2

DX1 , DX2 ? 10,15

Media de las  $Y$  comprendida entre MY1 y MY2

MY1 , MY2 ? 160,170

Desv. típ. de las  $Y$  comprendida entre DY1 y DY2

DY1 , DY2 ? 10,20

Coeficiente de correlación comprendido entre R1 y R2

R1 , R2? 0.8,0.9

Número de cifras enteras ( 1 , 2 , 3 , 4 ó 5 )? 3

Número de cifras decimales ( 0 , 1 ó 2 )? 0



Num. 1		Num. 2		Num. 3		Num. 4	
52	141	74	182	38	155	71	167
62	154	52	157	68	168	89	177
72	168	79	177	79	182	46	131
31	95	51	170	45	140	72	166
56	150	74	185	58	157	77	165
69	160	81	179	78	185	46	142
56	157	72	172	79	175	71	163
71	170	74	173	48	139	74	167
60	160	74	182	77	180	69	165
50	142	71	182	48	154	67	156
67	180	65	173	53	150	57	147
72	153	70	177	78	178	63	156
64	180	34	142	84	184	87	174
70	165	86	182	45	142	76	171
77	204	97	193	53	151	81	165
72	175	54	178	55	150	78	174
81	177	59	171	51	141	66	170
51	137	76	172	42	145	84	180
62	173	51	151	66	172	67	156
73	188	75	174	58	157	67	169

Num. 5		Num. 6		Num. 7		Num. 8	
77	182	75	175	68	161	44	144
61	170	58	152	46	134	77	182
70	166	69	175	63	153	53	150
68	170	61	150	87	182	57	165
67	156	50	144	50	142	53	150
53	131	82	183	75	175	65	169
79	174	51	147	84	185	56	159
67	160	62	157	66	152	52	161
69	170	70	170	76	171	51	156
72	164	76	179	71	181	89	184
88	171	63	162	72	163	83	197
66	178	93	178	79	175	70	185
84	179	61	153	69	179	72	178
39	154	70	160	67	161	77	183
71	152	76	178	70	174	63	164
52	146	80	184	80	178	64	169
61	158	81	172	74	167	81	184
57	141	48	149	73	167	66	170
41	129	59	150	80	172	78	184
64	158	66	162	59	161	58	164

\*\*\*\*\* S O L U C I O N E S \*\*\*\*\*

Num. 1

MEDIA de las X = 67      DESV.TIP. de las X = 13

MEDIA de las Y =165      DESV.TIP. de las Y = 20

Coeficiente de correlación = .86

Num. 2

MEDIA de las X = 64      DESV.TIP. de las X = 13

MEDIA de las Y =169      DESV.TIP. de las Y = 12

Coeficiente de correlación = .87

Num. 3

MEDIA de las X = 61      DESV.TIP. de las X = 13

MEDIA de las Y =163      DESV.TIP. de las Y = 15

Coeficiente de correlación = .87

Num. 4.

MEDIA de las X = 69      DESV.TIP. de las X = 13

MEDIA de las Y =161      DESV.TIP. de las Y = 13

Coeficiente de correlación = .85

Num. 5

MEDIA de las X = 65      DESV.TIP. de las X = 13

MEDIA de las Y =160      DESV.TIP. de las Y = 16

Coeficiente de correlación = .82

Num. 6

MEDIA de las X = 64      DESV.TIP. de las X = 15

MEDIA de las Y =162      DESV.TIP. de las Y = 16

Coeficiente de correlación = .89

Num. 7

MEDIA de las X = 70      DESV.TIP. de las X = 11

MEDIA de las Y =168      DESV.TIP. de las Y = 15

Coeficiente de correlación = .9

Num. 8

MEDIA de las X = 64      DESV.TIP. de las X = 12

MEDIA de las Y =168      DESV.TIP. de las Y = 13

Coeficiente de correlación = .88

```

10 REM Distribución bidimensional normal
20 CLS
30 PRINT "DISTRIBUCION BIDIMENSIONAL NORMAL"
40 DEFSNG A-Z
50 PRINT : PRINT
60 INPUT "Número de ejercicios";EJ%
70 DIM MX(EJ%),DX(EJ%),MY(EJ%),DY(EJ%),R(EJ%)
80 U$(0)="" : U$(1)="#" : U$(2)="##" : U$(3)="###"
  : U$(4)="####":U$(5)="#####
90 PRINT : INPUT "Número de datos en cada ejercicio ( 20 ó 50 )";DA%
100 PRINT : PRINT "Media de las x comprendida entre MX1 y MX2" : PRINT
110 INPUT "MX1 , MX2 ";M1,M2
120 PRINT : PRINT "Desv. típ. de las x comprendida entre DX1 y DX2" : PRINT
130 INPUT "DX1 , DX2 ";D1,D2
140 PRINT : PRINT "Media de las Y comprendida entre MY1 y MY2" : PRINT
150 INPUT "MY1 , MY2 ";M3,M4
160 PRINT : PRINT "Desv. típ. de las Y comprendida entre DY1 y DY2" : PRINT
170 INPUT "DY1 , DY2 ";D3,D4
180 PRINT : PRINT "Coeficiente de correlación comprendido entre R1 Y R2" : PRINT
190 INPUT "R1 , R2";R1,R2
200 PRINT : INPUT "Número de cifras enteras ( 1 , 2 , 3 , 4 ó 5 )";CE%
210 PRINT : INPUT "Número de cifras decimales ( 0 , 1 ó 2 )";CD%
220 U$=U$(CE%)+". "+U$(CD%)
230 RE%=10^CD% : DEF FN RED(Z)=INT(RE%*Z+.5)/RE%
240 FOR K%=1 TO EJ%
250 PRINT : PRINT : PRINT "Num.";K% : PRINT
260 MX(K%)=M1+(M2-M1)*RND(1) : MX(K%)=FN RED(MX(K%))
270 DX(K%)=D1+(D2-D1)*RND(1) : DX(K%)=FN RED(DX(K%))
280 MY(K%)=M3+(M4-M3)*RND(1) : MY(K%)=FN RED(MY(K%))
290 DY(K%)=D3+(D4-D3)*RND(1) : DY(K%)=FN RED(DY(K%))
300 R(K%)=R1+(R2-R1)*RND(1)

```

```

310 FOR I%=1 TO DA%
320 GOSUB 550
330 Z1=Z
340 GOSUB 550
350 Z2=Z
360 X=MX(K%)+DX(K%)*Z1
370 Y=MY(K%)+DY(K%)*(R(K%)*Z1+SQR(1-R(K%)*R(K%))*
Z2)
380 PRINT USING U$;X; : PRINT " "; : PRINT USING
U$;Y
390 NEXT I%
400 NEXT K%
410 PRINT : PRINT
420 REM Escritura de las soluciones
430 PRINT "***** S O L U C I O N E S *****"
440 PRINT : PRINT
450 FOR K%=1 TO EJ%
460 PRINT "Num.";K%
470 PRINT "MEDIA de las X ="; : PRINT USING U$;MX(
K%);
480 PRINT "    DESV.TIP. de las X ="; : PRINT USING
U$;DX(K%)
490 PRINT "MEDIA de las Y ="; : PRINT USING U$;MY(
K%);
500 PRINT "    DESV.TIP. de las Y ="; : PRINT USING
U$;DY(K%)
510 PRINT "Coeficiente de correlación =";INT(R(K%)
*100+.5)/100
520 PRINT
530 NEXT K%
540 END
550 REM Subrutina N(0,1)
560 S=0
570 FOR H%=1 TO 12
580 S=S+RND(1)
590 NEXT H%
600 Z=S-6
610 RETURN

```

Guarde este programa con el nombre de DISTRIBI.BAS.

# 9

## RESOLUCION DE TRIANGULOS

### 9.1 TRIANGULOS RECTANGULOS

Cada ejercicio de este programa propone la resolución de cuatro casos de triángulos rectángulos  $ABC$  ( $A = 90^\circ$ ).

El lector puede obtener tantos ejercicios como desee ejecutando el programa:

T R I A N G U L O S   R E C T A G U L O S

Cuántos ejercicios? 10

Num. 1 Resuelve los triángulos rectángulos :

- |      |         |                |
|------|---------|----------------|
| 1.1) | $b = 7$ | $c = 6$        |
| 1.2) | $a = 6$ | $b = 3$        |
| 1.3) | $c = 7$ | $B = 28^\circ$ |
| 1.4) | $a = 5$ | $B = 42^\circ$ |



Num. 2 Resuelve los triángulos rectángulos :

- |      |         |                |
|------|---------|----------------|
| 2.1) | $b = 4$ | $c = 7$        |
| 2.2) | $a = 5$ | $b = 4$        |
| 2.3) | $c = 5$ | $B = 51^\circ$ |
| 2.4) | $a = 6$ | $B = 38^\circ$ |



Num. 3 Resuelve los triángulos rectángulos :

- 3.1)  $b = 4$   $c = 7$   
 3.2)  $a = 4$   $b = 2$   
 3.3)  $c = 6$   $B = 41^\circ$   
 3.4)  $a = 4$   $B = 35^\circ$



Num. 4 Resuelve los triángulos rectángulos :

- 4.1)  $b = 8$   $c = 7$   
 4.2)  $a = 6$   $b = 3$   
 4.3)  $c = 4$   $B = 43^\circ$   
 4.4)  $a = 6$   $B = 48^\circ$



Num. 5 Resuelve los triángulos rectángulos :

- 5.1)  $b = 4$   $c = 5$   
 5.2)  $a = 8$   $b = 5$   
 5.3)  $c = 4$   $B = 40^\circ$   
 5.4)  $a = 4$   $B = 30^\circ$



Num. 6 Resuelve los triángulos rectángulos :

- 6.1)  $b = 5$   $c = 4$   
 6.2)  $a = 8$   $b = 7$   
 6.3)  $c = 6$   $B = 50^\circ$   
 6.4)  $a = 7$   $B = 28^\circ$




Num. 7 Resuelve los triángulos rectángulos :

- 7.1)  $b = 4$   $c = 7$   
 7.2)  $a = 4$   $b = 3$   
 7.3)  $c = 7$   $B = 27^\circ$   
 7.4)  $a = 6$   $B = 27^\circ$



Num. 8 Resuelve los triángulos rectángulos :

- 8.1)  $b = 7$   $c = 5$   
 8.2)  $a = 7$   $b = 4$   
 8.3)  $c = 8$   $B = 45^\circ$   
 8.4)  $a = 5$   $B = 41^\circ$



Num. 9 Resuelve los triángulos rectángulos :

- 9.1)  $b = 7$   $c = 6$   
 9.2)  $a = 6$   $b = 3$   
 9.3)  $c = 8$   $B = 30^\circ$   
 9.4)  $a = 5$   $B = 26^\circ$



Num. 10 Resuelve los triángulos rectángulos :

- 10.1)  $b = 7$   $c = 3$   
 10.2)  $a = 8$   $b = 7$   
 10.3)  $c = 4$   $B = 27^\circ$   
 10.4)  $a = 4$   $B = 26^\circ$



## SOLUCIONES DE TRIANGULOS RECTANGULOS

Num. 1

- 1.1)  $a = 9.22$   $B = 49^\circ 23' 55''$   $C = 40^\circ 36' 5''$   
 1.2)  $c = 5.20$   $B = 30^\circ 0' 0''$   $C = 60^\circ 0' 0''$   
 1.3)  $a = 7.93$   $b = 3.72$   $C = 62^\circ$   
 1.4)  $b = 3.35$   $c = 3.72$   $C = 48^\circ$

Num. 2

- 2.1)  $a = 8.06$   $B = 29^\circ 44' 42''$   $C = 60^\circ 15' 18''$   
 2.2)  $c = 3.00$   $B = 53^\circ 7' 48''$   $C = 36^\circ 52' 12''$   
 2.3)  $a = 7.95$   $b = 6.17$   $C = 39^\circ$   
 2.4)  $b = 3.69$   $c = 4.73$   $C = 52^\circ$

### Num. 3

3.1)	a = 8.06	B = 29° 44' 42''	C = 60° 15' 18''
3.2)	c = 3.46	B = 30° 0' 0''	C = 60° 0' 0''
3.3)	a = 7.95	b = 5.22	C = 49°
3.4)	b = 2.29	c = 3.28	C = 55°

### Num. 4

4.1)	a = 10.63	B = 48° 48' 51''	C = 41° 11' 9''
4.2)	c = 5.20	B = 30° 0' 0''	C = 60° 0' 0''
4.3)	a = 5.47	b = 3.73	C = 47°
4.4)	b = 4.46	c = 4.01	C = 42°

### Num. 5

5.1)	a = 6.40	B = 38° 39' 35''	C = 51° 20' 25''
5.2)	c = 6.24	B = 38° 40' 56''	C = 51° 19' 4''
5.3)	a = 5.22	b = 3.36	C = 50°
5.4)	b = 2.00	c = 3.46	C = 60°

### Num. 6

6.1)	a = 6.40	B = 51° 20' 25''	C = 38° 39' 35''
6.2)	c = 3.87	B = 61° 2' 42''	C = 28° 57' 18''
6.3)	a = 9.33	b = 7.15	C = 40°
6.4)	b = 3.29	c = 6.18	C = 62°

### Num. 7

7.1)	a = 8.06	B = 29° 44' 42''	C = 60° 15' 18''
7.2)	c = 2.65	B = 48° 35' 25''	C = 41° 24' 35''
7.3)	a = 7.86	b = 3.57	C = 63°
7.4)	b = 2.72	c = 5.35	C = 63°

### Num. 8

8.1)	a = 8.60	B = 54° 27' 44''	C = 35° 32' 16''
8.2)	c = 5.74	B = 34° 51' 0''	C = 55° 9' 0''
8.3)	a = 11.31	b = 8.00	C = 45°
8.4)	b = 3.28	c = 3.77	C = 49°

### Num. 9

9.1)	a = 9.22	B = 49° 23' 55''	C = 40° 36' 5''
9.2)	c = 5.20	B = 30° 0' 0''	C = 60° 0' 0''
9.3)	a = 9.24	b = 4.62	C = 60°
9.4)	b = 2.19	c = 4.49	C = 64°

### Num. 10

10.1)	a = 7.62	B = 66° 48' 5''	C = 23° 11' 55''
10.2)	c = 3.87	B = 61° 2' 42''	C = 28° 57' 18''
10.3)	a = 4.49	b = 2.04	C = 63°
10.4)	b = 1.75	c = 3.60	C = 64°



Este es el programa sobre triángulos rectángulos:

```
10 REM Resolución de triángulos rectángulos
20 CLS
30 PRINT "T R I A N G U L O S   R E C T A G U L O S"
40 DEFSNG A-Z
50 PRINT : PRINT
60 INPUT "Cuántos ejercicios";EJ
70 DIM LA(EJ,4),LB(EJ,2),LC(EJ,3),GB(EJ,4)
80 PI=4*ATN(1) : G$=CHR$(248)
90 FOR K=1 TO EJ
100 PRINT : PRINT
110 PRINT"Num.;"K ;"Resuelve los triángulos rectán
gulos :" : PRINT
120 REM Caso 1
130 LB(K,1)=INT(RND(1)*5+4)
140 LC(K,1)=INT(RND(1)*5+3)
150 PRINT TAB(37)"B"
160 PRINT USING "###.##";K+.1;
170 PRINT " )   b=";LB(K,1);"   c=";LC(K,1);TAB(36)"/i"
180 REM Caso 2
190 LA(K,2)=INT(RND(1)*5+4)
200 LB(K,2)=LA(K,2)-INT(RND(1)*3+1)
210 PRINT USING "###.##";K+.2;
220 PRINT " )       a=";LA(K,2);"       b=";LB(K,2);TAB(33
)"a / i c"
230 REM Caso 3
240 LC(K,3)=INT(RND(1)*5+4)
250 GB(K,3)=INT(RND(1)*30+25)
260 PRINT USING "###.##";K+.3;
270 PRINT " )       c=";LC(K,3);"       B=";STR$(GB(K,3))
;G$;
280 PRINT TAB(34)"/ i"
290 REM Caso 4
300 LA(K,4)=INT(RND(1)*5+4)
310 GB(K,4)=INT(RND(1)*30+25)
320 PRINT USING "###.##";K+.4;
330 PRINT " )       a=";LA(K,4);"       B=";STR$(GB(K,4))
;G$;
```

```

340 PRINT TAB(33)"/____i"
350 PRINT TAB(32)"C  b  A"
360 NEXT K
370 PRINT : PRINT : PRINT
380 PRINT "SOLUCIONES DE TRIANGULOS RECTANGULOS"
390 FOR K=1 TO EJ
400 PRINT
410 PRINT"Num.";K
420 REM Caso 1
430 LA=SQR(LB(K,1)^2+LC(K,1)^2)
440 RX=ATN(LB(K,1)/LC(K,1)) : GOSUB 820
450 GB=GX : MB=MX : SB= SX
460 RX=PI/2-RX : GOSUB 820
470 GC=GX : MC=MX : SC= SX
480 PRINT USING "###.##";K+.1;
490 PRINT ")  a=";USING "##.##";LA;
500 PRINT TAB(19)"B=";STR$(GB);G$;STR$(MB);"";STR$(
  SB);"";
510 PRINT TAB(35)"C=";STR$(GC);G$;STR$(MC);"";STR$(
  SC);""
520 REM Caso 2
530 LC=SQR(LA(K,2)^2-LB(K,2)^2)
540 RX=ATN(LB(K,2)/LC) : GOSUB 820
550 GB=GX : MB=MX : SB= SX
560 RX=PI/2-RX : GOSUB 820
570 GC=GX : MC=MX : SC= SX
580 PRINT USING "###.##";K+.2;
590 PRINT ")  c=";USING "##.##";LC;
600 PRINT TAB(19)"B=";STR$(GB);G$;STR$(MB);"";STR$(
  SB);"";
610 PRINT TAB(35)"C=";STR$(GC);G$;STR$(MC);"";STR$(
  SC);""
620 REM Caso 3
630 RB=PI/180*GB(K,3)
640 LB=LC(K,3)*TAN(RB)
650 LA=LC(K,3)/COS(RB)
660 GC=90-GB(K,3)
670 PRINT USING "###.##";K+.3;
680 PRINT ")  a=";USING "##.##";LA;

```

```

690 PRINT TAB(19)"b="; USING "##.##"; LB;
700 PRINT TAB(35)"C="; STR$(GC); G$
710 REM Caso 4
720 RB=PI/180*GB(K,4)
730 LC=LA(K,4)*COS(RB)
740 LB=LA(K,4)*SIN(RB)
750 GC=90-GB(K,4)
760 PRINT USING "###.##"; K+.4;
770 PRINT " )  b="; USING "##.##"; LB;
780 PRINT TAB(19)"c="; USING "##.##"; LC;
790 PRINT TAB(35)"C="; STR$(GC); G$
800 NEXT K
810 END
820 REM Subrutina RADIAN --> GRA,MIN,SEG
830 DX=RX*180/PI
840 GX=INT(DX)
850 MX=INT(60*(DX-GX))
860 SX=INT(60*(60*(DX-GX)-MX)+.5)
870 IF SX=60 THEN SX=0 : MX=MX+1
880 IF MX=60 THEN MX=0 : GX=GX+1
890 RETURN

```

Si su pantalla es de 40 caracteres por línea y quiere que los resultados salgan en una sola línea, deberá reorganizar las instrucciones PRINT TAB y renunciar a que figuren los segundos de ángulo.

Guarde el programa con el nombre de TRIANREC.BAS.

## 9.2 TRIANGULOS CUALESQUIERA

Este programa es análogo al anterior: propone la resolución de los cuatro casos. En el caso tercero los datos siempre han sido elegidos de modo que haya dos soluciones.

Veamos cómo funciona el programa:

**T R I A N G U L O S**

Cuántos ejercicios? 10

Num. 1 Resuelve los siguientes triángulos :

- 1.1)  $c = 7$      $A = 50^\circ$      $B = 84^\circ$
- 1.2)  $b = 8$      $c = 7$      $A = 45^\circ$

- 1.3)  $a = 4$      $c = 5$      $A = 30^\circ$   
 1.4)  $a = 4$      $b = 8$      $c = 9$

Num. 2 Resuelve los siguientes triángulos :

- 2.1)  $c = 5$      $A = 39^\circ$      $B = 76^\circ$   
 2.2)  $b = 8$      $c = 6$      $A = 85^\circ$   
 2.3)  $a = 3$      $c = 4$      $A = 33^\circ$   
 2.4)  $a = 4$      $b = 5$      $c = 6$

Num. 3 Resuelve los siguientes triángulos :

- 3.1)  $c = 6$      $A = 46^\circ$      $B = 73^\circ$   
 3.2)  $b = 5$      $c = 8$      $A = 134^\circ$   
 3.3)  $a = 4$      $c = 6$      $A = 32^\circ$   
 3.4)  $a = 4$      $b = 7$      $c = 8$

Num. 4 Resuelve los siguientes triángulos :

- 4.1)  $c = 6$      $A = 53^\circ$      $B = 72^\circ$   
 4.2)  $b = 6$      $c = 8$      $A = 128^\circ$   
 4.3)  $a = 3$      $c = 4$      $A = 30^\circ$   
 4.4)  $a = 4$      $b = 4$      $c = 5$

Num. 5 Resuelve los siguientes triángulos :

- 5.1)  $c = 5$      $A = 41^\circ$      $B = 97^\circ$   
 5.2)  $b = 4$      $c = 6$      $A = 133^\circ$   
 5.3)  $a = 5$      $c = 7$      $A = 26^\circ$   
 5.4)  $a = 4$      $b = 8$      $c = 9$

Num. 6 Resuelve los siguientes triángulos :

- 6.1)  $c = 4$      $A = 39^\circ$      $B = 89^\circ$   
 6.2)  $b = 4$      $c = 6$      $A = 40^\circ$   
 6.3)  $a = 5$      $c = 7$      $A = 29^\circ$   
 6.4)  $a = 7$      $b = 7$      $c = 11$

Num. 7 Resuelve los siguientes triángulos :

- 7.1)  $c = 8$      $A = 50^\circ$      $B = 76^\circ$   
7.2)  $b = 6$      $c = 7$      $A = 109^\circ$   
7.3)  $a = 4$      $c = 6$      $A = 33^\circ$   
7.4)  $a = 8$      $b = 4$      $c = 9$

Num. 8 Resuelve los siguientes triángulos :

- 8.1)  $c = 5$      $A = 31^\circ$      $B = 92^\circ$   
8.2)  $b = 4$      $c = 8$      $A = 69^\circ$   
8.3)  $a = 3$      $c = 4$      $A = 25^\circ$   
8.4)  $a = 4$      $b = 4$      $c = 5$

Num. 9 Resuelve los siguientes triángulos :

- 9.1)  $c = 5$      $A = 32^\circ$      $B = 79^\circ$   
9.2)  $b = 6$      $c = 4$      $A = 132^\circ$   
9.3)  $a = 4$      $c = 5$      $A = 25^\circ$   
9.4)  $a = 7$      $b = 5$      $c = 9$

Num. 10 Resuelve los siguientes triángulos :

- 10.1)  $c = 5$      $A = 59^\circ$      $B = 75^\circ$   
10.2)  $b = 4$      $c = 5$      $A = 101^\circ$   
10.3)  $a = 5$      $c = 7$      $A = 25^\circ$   
10.4)  $a = 6$      $b = 5$      $c = 8$

\*\*\* SOLUCIONES DE TRIANGULOS \*\*\*

Num. 1

- |      |                        |                         |                         |
|------|------------------------|-------------------------|-------------------------|
| 1.1) | $a = 7.45$             | $b = 9.68$              | $C = 46^\circ$          |
| 1.2) | $a = 5.81$             | $B = 76^\circ 38' 36''$ | $C = 58^\circ 21' 24''$ |
| 1.3) | $b = 7.45$             | $B = 111^\circ 19' 4''$ | $C = 38^\circ 40' 56''$ |
| 1.3) | $b = 1.21$             | $B = 8^\circ 40' 56''$  | $C = 141^\circ 19' 4''$ |
| 1.4) | $A = 26^\circ 23' 4''$ | $B = 62^\circ 43' 13''$ | $C = 90^\circ 53' 43''$ |

Num. 2

- |      |                         |                          |                          |
|------|-------------------------|--------------------------|--------------------------|
| 2.1) | $a = 3.47$              | $b = 5.35$               | $C = 65^\circ$           |
| 2.2) | $a = 9.57$              | $B = 56^\circ 21' 40''$  | $C = 38^\circ 38' 20''$  |
| 2.3) | $b = 5.42$              | $B = 100^\circ 25' 57''$ | $C = 46^\circ 34' 3''$   |
| 2.3) | $b = 1.29$              | $B = 13^\circ 34' 3''$   | $C = 133^\circ 25' 57''$ |
| 2.4) | $A = 41^\circ 24' 35''$ | $B = 55^\circ 46' 16''$  | $C = 82^\circ 49' 9''$   |

Num. 3

3.1)	a= 4.93	b= 6.56	C= 61°
3.2)	a=12.02	B= 17° 24' 19''	C= 28° 35' 41''
3.3)	b= 7.52	B= 95° 21' 22''	C= 52° 38' 38''
3.3)	b= 2.66	B= 20° 38' 38''	C= 127° 21' 22''
3.4)	A= 29° 59' 41''	B= 61° 1' 42''	C= 88° 58' 36''

Num. 4

4.1)	a= 5.85	b= 6.97	C= 55°
4.2)	a=12.61	B= 22° 0' 51''	C= 29° 59' 9''
4.3)	b= 5.70	B= 108° 11' 23''	C= 41° 48' 37''
4.3)	b= 1.23	B= 11° 48' 37''	C= 138° 11' 23''
4.4)	A= 51° 19' 4''	B= 51° 19' 4''	C= 77° 21' 52''

Num. 5

5.1)	a= 4.90	b= 7.42	C= 42°
5.2)	a= 9.21	B= 18° 31' 48''	C= 28° 28' 12''
5.3)	b=10.24	B= 116° 8' 28''	C= 37° 51' 32''
5.3)	b= 2.34	B= 11° 51' 32''	C= 142° 8' 28''
5.4)	A= 26° 23' 4''	B= 62° 43' 13''	C= 90° 53' 43''

Num. 6

6.1)	a= 3.19	b= 5.08	C= 52°
6.2)	a= 3.90	B= 41° 12' 41''	C= 98° 47' 19''
6.3)	b= 9.79	B= 108° 15' 19''	C= 42° 44' 41''
6.3)	b= 2.45	B= 13° 44' 41''	C= 137° 15' 19''
6.4)	A= 38° 12' 48''	B= 38° 12' 48''	C= 103° 34' 25''

Num. 7

7.1)	a= 7.58	b= 9.59	C= 54°
7.2)	a=10.60	B= 32° 21' 34''	C= 38° 38' 26''
7.3)	b= 7.34	B= 92° 13' 7''	C= 54° 46' 53''
7.3)	b= 2.73	B= 21° 46' 53''	C= 125° 13' 7''
7.4)	A= 62° 43' 13''	B= 26° 23' 4''	C= 90° 53' 43''

Num. 8

8.1)	a= 3.07	b= 5.96	C= 57°
8.2)	a= 7.55	B= 29° 37' 35''	C= 81° 22' 25''
8.3)	b= 6.10	B= 120° 42' 9''	C= 34° 17' 51''
8.3)	b= 1.15	B= 9° 17' 51''	C= 145° 42' 9''
8.4)	A= 51° 19' 4''	B= 51° 19' 4''	C= 77° 21' 52''

Num. 9

9.1)	a= 2.84	b= 5.26	C= 69°
9.2)	a= 9.17	B= 29° 5' 19''	C= 18° 54' 41''
9.3)	b= 7.93	B= 123° 6' 40''	C= 31° 53' 20''
9.3)	b= 1.14	B= 6° 53' 20''	C= 148° 6' 40''
9.4)	A= 50° 42' 13''	B= 33° 33' 26''	C= 95° 44' 21''

Num. 10

10.1)	a= 5.96	b= 6.71	C= 46°
10.2)	a= 6.97	B= 34° 16' 0''	C= 44° 44' 0''
10.3)	b=10.38	B= 118° 43' 29''	C= 36° 16' 31''
10.3)	b= 2.31	B= 11° 16' 31''	C= 143° 43' 29''
10.4)	A= 48° 30' 33''	B= 38° 37' 29''	C= 92° 51' 58''

He aquí el programa que propone ejercicios de triángulos:

```
10 REM Resolución de triángulos
20 CLS
30 PRINT "T R I A N G U L O S" : PRINT : PRINT
40 DEFSNG A-Z
50 INPUT "Cuántos ejercicios";EJ
60 DIM LA(EJ,4),LB(EJ,4),LC(EJ,4),GA(EJ,4),GB(EJ,4)
70 PI=4*ATN(1) : G$=CHR$(248)
80 REM arco coseno ( x no 0 )
90 DEF FN AR(X)=ATN(SQR(1-X*X)/X)+(1-SGN(X))/2*PI
100 FOR K=1 TO EJ
110 PRINT : PRINT
120 PRINT"Num. ";K ; "Resuelve los siguientes triángulos : " : PRINT
130 REM Caso 1
140 LC(K,1)=INT(RND(1)*5+4)
150 GA(K,1)=INT(RND(1)*30+30)
160 GB(K,1)=INT(RND(1)*30+70)
170 PRINT USING "###.##";K+.1;
180 PRINT " ) c=";LC(K,1); " A=";STR$(GA(K,1));G$;
" B=";STR$(GB(K,1));G$
190 REM Caso 2
200 LB(K,2)=INT(RND(1)*5+4)
210 LC(K,2)=INT(RND(1)*5+4)
220 GA(K,2)=INT(RND(1)*120+30)
230 PRINT USING "###.##";K+.2;
240 PRINT " ) b=";LB(K,2); " c=";LC(K,2); " A=";
STR$(GA(K,2));G$
250 REM Caso 3
260 LC(K,3)=INT(RND(1)*5+4)
270 LA(K,3)=INT(LC(K,3)*.8)
280 GA(K,3)=INT(RND(1)*10+25)
290 PRINT USING "###.##";K+.3;
```

```

300 PRINT " ) a=";LA(K,3);" c=";LC(K,3);" A=";
STR$(GA(K,3));G$
310 REM Caso 4
320 LA(K,4)=INT(RND(1)*5+4)
330 LB(K,4)=INT(RND(1)*5+4)
340 LC(K,4)=LA(K,4)+LB(K,4)-3
350 PRINT USING "###.##";K+.4;
360 PRINT " ) a=";LA(K,4);" b=";LB(K,4);" c=";
LC(K,4)
370 NEXT K
380 PRINT : PRINT : PRINT
390 PRINT "*** SOLUCIONES DE TRIANGULOS ***"
400 FOR K=1 TO EJ
410 PRINT
420 PRINT"Num. ";K
430 REM Caso 1
440 GC=180-GA(K,1)-GB(K,1)
450 RA=GA(K,1)/180*PI
460 RB=GB(K,1)/180*PI
470 RC=PI-RA-RB
480 LA=LC(K,1)*SIN(RA)/SIN(RC)
490 LB=LC(K,1)*SIN(RB)/SIN(RC)
500 PRINT USING "###.##";K+.1;
510 PRINT " ) a=";USING "##.##";LA;
520 PRINT TAB(25)"b=";USING "##.##";LB;
530 PRINT TAB(42)"c=";STR$(GC);G$
540 REM Caso 2
550 RA=GA(K,2)/180*PI
560 LA=SQR(LB(K,2)^2+LC(K,2)^2-2*LB(K,2)*LC(K,2)*
COS(RA))
570 CB=(LA^2+LC(K,2)^2-LB(K,2)^2)/(2*LA*LC(K,2))
580 IF CB=0 THEN RB=PI/2 ELSE RB=FNAR(CB)
590 RC=PI-RA-RB
600 RX=RB : GOSUB 1100
610 GB=GX : MB=MX : SB= SX
620 RX=RC : GOSUB 1100
630 GC=GX : MC=MX : SC= SX
640 PRINT USING "###.##";K+.2;
650 PRINT " ) a=";USING "##.##";LA;

```



```

660 PRINT TAB(25) "B="; STR$(GB); G$; STR$(MB); " ";
STR$(SB); " ";
670 PRINT TAB(42) "C="; STR$(GC); G$; STR$(MC); " ";
STR$(SC); " ";
680 REM Caso 3
690 RA=GA(K,3)/180*PI
700 DI=LC(K,3)*SIN(RA)
710 SC=DI/LA(K,3) : CC=SQR(1-SC^2)
720 IF CC=0 THEN C1=PI/2 : C2=C1 : GOTO 740
730 C1=FN AR(CC) : C2=FN AR(-CC)
740 B1=PI-C1-RA : B2=PI-C2-RA
750 L1=SQR(LA(K,3)^2+LC(K,3)^2-2*LA(K,3)*LC(K,3)*
COS(B1))
760 L2=SQR(LA(K,3)^2+LC(K,3)^2-2*LA(K,3)*LC(K,3)*
COS(B2))
770 PRINT USING "###.##"; K+.3;
780 PRINT ") b="; USING "##.##"; L1;
790 RX=B1; GOSUB 1100
800 PRINT TAB(25) "B="; STR$(GX); G$; STR$(MX); " ";
STR$(SX); " ";
810 RX=C1 : GOSUB 1100
820 PRINT TAB(42) "C="; STR$(GX); G$; STR$(MX); " ";
STR$(SX); " ";
830 IF LC(K,3) <= LA(K,3) THEN RETURN
840 PRINT USING "###.##"; K+.3;
850 PRINT ") b="; USING "##.##"; L2;
860 RX=B2 : GOSUB 1100
870 PRINT TAB(25) "B="; STR$(GX); G$; STR$(MX); " ";
STR$(SX); " ";
880 RX=C2 : GOSUB 1100
890 PRINT TAB(42) "C="; STR$(GX); G$; STR$(MX); " ";
STR$(SX); " ";
900 REM Caso 4
910 LA=LA(K,4) : LB=LB(K,4) : LC=LC(K,4)
920 CA=(LB^2+LC^2-LA^2)/(2*LB*LC)
930 CB=(LA^2+LC^2-LB^2)/(2*LA*LC)
940 CC=(LA^2+LB^2-LC^2)/(2*LA*LB)
950 RA=FN AR(CA)
960 RB=FN AR(CB)

```

```

970 RC=FN AR(CC)
980 RX=RA : GOSUB 1100
990 GA=GX : MA=MX : SA=SX
1000 PRINT USING "###.##";K+.4;
1010 PRINT " ) A=";STR$(GA);G$;STR$(MA);"°";
STR$(SA);"°°";
1020 RX=RB : GOSUB 1100
1030 GB=GX : MB=MX : SB=SX
1040 PRINT TAB(25)"B=";STR$(GB);G$;STR$(MB);"°";
STR$(SB);"°°";
1050 RX=RC : GOSUB 1100
1060 GC=GX : MC=MX : SC=SX
1070 PRINT TAB(42)"C=";STR$(GC);G$;STR$(MC);"°";
STR$(SC);"°°"
1080 NEXT K
1090 END
1100 REM Subrutina RADIAN --> GRA,MIN,SEG
1110 DX=RX*180/PI
1120 GX=INT(DX)
1130 MX=INT(60*(DX-GX))
1140 SX=INT(60*(60*(DX-GX)-MX)+.5)
1150 IF SX=60 THEN SX=0 : MX=MX+1
1160 IF MX=60 THEN MX=0 : GX=GX+1
1170 RETURN

```

Guarde el programa con el nombre de TRIANGUL.BAS.

# 10

## REPRESENTACION GRAFICA DE FUNCIONES

### 10.1. REPRESENTACION GRAFICA DE FUNCIONES

En este capítulo se rompe el esquema que hemos seguido en todos los programas del libro, a saber: primero presentar en la pantalla, o en el papel de la impresora los enunciados de los ejercicios generados por el ordenador, y después, las soluciones correspondientes.

En el caso de la *representación gráfica de funciones* es muy difícil conseguir que el ordenador genere enunciados, es decir, definiciones de funciones, a no ser que nos limitemos a un tipo determinado de ellas y variemos las funciones con la adición (o multiplicación), de constantes aleatorias, cosa que no tiene demasiado interés. Parece más sensato ofrecer un programa flexible para representar cualquier función que se pueda definir con el lenguaje BASIC.

Veamos cómo funciona el programa:

```
REPRESENTACION GRAFICA DE UNA FUNCION F(X)
```

```
Complete la definición de la función f(x) de la línea 120 y pulse la tecla  
< RETURN >.
```

```
Escriba RUN 70 y pulse la tecla < RETURN >.
```

```
120 DEF FN F(X)=X*SIN(X)
```

```
Ok
```

```
run 70
```

```
Extremos del intervalo (a,b)? -11,11
```

```
E S T O Y   T R A B A J A N D O
```

```
Espera unos segundos
```

Al cabo de un tiempo aparece en la pantalla esta gráfica:




figura 1

Nótese que en la primera línea de la pantalla aparece el intervalo  $[a,b]$ , así como el valor máximo  $MA$  y el valor mínimo  $mi$  que alcanza la función en dicho intervalo.

El programa aprovecha la pantalla lo mejor posible: sitúa el punto más alto de la gráfica en el borde superior, el más bajo en el borde inferior y los ejes en el sitio que les corresponde.

Con este programa es posible estudiar con más detalle un trozo determinado de curva (poner la lupa). Basta para ello restringir el intervalo de definición. Así, si hacemos  $a = -6$  y  $b = 6$ , resulta la gráfica:




figura 2

Este es el programa de representaciones gráficas:

```

10 REM Representación gráfica de f(x)
20 CLS : KEY OFF
30 PRINT "R E P R E S E N T A C I O N   G R A F I
CA   D E   U N A   F U N C I O N   F ( X ) " : PRINT
40 PRINT "Complete la definición de la función
f(x) de la línea 120 y pulse la tecla" : PRINT
: PRINT "< RETURN >." : PRINT : PRINT
50 PRINT "Escriba RUN 70 y pulse la tecla <
RETURN >." : PRINT : PRINT
60 LIST 120
70 DEF SNG A-Z
80 PRINT : PRINT : INPUT "Extremos del intervalo

```

```

(a,b)";A,B
90 DIM Y(639)
100 CLS : LOCATE 12,21 : PRINT "E S T O Y   T R A B A
J A N D O"
110 LOCATE 18,26 : PRINT "Espere unos segundos"
120 DEF FN F(X)=X*SIN(X)
130 REM Cálculo del máximo y del mínimo
140 E=(B-A)/639
150 MA=FN F(B) : MI=MA
160 FOR I%=0 TO 639
170 Y(I%)=FN F(A+I%*E)
180 IF Y(I%)>MA THEN MA=Y(I%)
190 IF Y(I%)<MI THEN MI=Y(I%)
200 NEXT I%
210 CLS
220 REM Dibujo de los ejes
230 SCREEN 2
240 PRINT "      a=";A;"      b=";B;TAB(44)"Ma=";INT(
MA*100+.5)/100;"      mi=";INT(MI*100+.5)/100
250 D=199/(MA-MI)
260 LINE(0,D*MA)-(639,D*MA)
270 LINE(-A/E,0)-(-A/E,199)
280 REM Dibujo de la curva
290 FOR I%=0 TO 639
300 J%=-D*Y(I%)+D*MA
310 PSET(I%,J%)
320 NEXT I%
330 END

```

Guárdelo con el nombre de GRAFICAS.BAS.



La generación aleatoria de ejercicios es una potente herramienta para obtener un número ilimitado de ejercicios, junto con las soluciones correspondientes.

En este libro se presenta la teoría fundamental y ejemplos concretos de:

Divisibilidad.

Operaciones con fracciones.

Sistemas de numeración.

Ecuaciones.

Polinomios.

Determinantes y sistemas.

Estadística.

Resolución de triángulos.

Representación gráfica de funciones.

Una colección de libros básicos:

- **BASIC BASICO. CURSO DE PROGRAMACION**
- **PROGRAMAS COMENTADOS DE BASIC BASICO**
- **BASIC JUNIOR. INICIACION A LA PROGRAMACION**
- **PROGRAMAS EXPLICADOS DE BASIC JUNIOR**
- **EL BASIC del Spectrum. Del teclado al microdrive**